1400367932

MATEMÀTIQUES

Quadern de pràctiques amb *MAPLE V*

Part I

Montserrat Bruguera i Padró
Albert Ferrer i Biosca
Antoni Guillamon i Grabolosa
Margarida Mitjana i Riera
Francesc Panyella i Brustenga
Xavier Puerta i Coll
Joan J. Rodríguez i Jordana
Carles Serrat i Piè

E.U.P.B.

Els programes de manipulació algebraica (també anomenats de *manipulació simbòlica* o *sistemes algebraics per a ordinador*) estan dissenyats amb l'objectiu d'efectuar operacions de càlcul simbòlic, a més de servir com a suport gràfic i numèric i, en alguns casos, com a llenguatge de programació. Aquests programes poden estalviar-nos una enorme quantitat de temps i esforç dedicats al desenvolupament i simplificació d'expressions matemàtiques. D'altra banda, el tractament simbòlic enfront del tractament numèric és molt més fiable i permet una millor interpretació dels resultats.

Actualment, dos dels programes més utilitzats per les seves altes prestacions són *MATHEMATICA* i *MAPLE V*. Com és de suposar, també són molt exigents amb les característiques dels ordinadors i la preparació prèvia dels seus usuaris. De tota manera, la majoria d'ordinadors personals que avui en dia es comercialitzen són capaços de suportar aquestes demandes tècniques. Quant a la preparació dels usuaris, es tracta de familiaritzar-se amb la sintaxi que s'hi utilitza i incorporar gradualment el coneixement i l'ús de noves comandes.

Amb aquest quadern pretenem introduir l'alumne en l'ús de manipuladors algebraics, donant les pautes més elementals d'utilització del programa *MAPLE V*, així com indicacions per a la seva utilització en els temes vinculats als cursos de Matemàtiques dels estudis d'Arquitectura Tècnica i Enginyeria Tècnica en Topografia que s'imparteixen a la Escola Universitària Politècnica de Barcelona. El seu contingut està pensat bàsicament per ser utilitzat en les sessions pràctiques d'aquestes assignatures.

Hem triat *MAPLE V* perquè, dins del ventall de manipuladors algebraics, és dels programes més complets, i perquè, per les seves característiques, s'adapta fàcilment a l'ús que se n'hagi de fer. Les llibreries contenen informació específica *linalg, plots, ...* que només es llegeix en el moment que cal, de forma que, en general, queda el màxim de memòria disponible per a executar els processos sol.licitats per l'usuari. Les prestacions gràfiques són molt elevades i es desenvolupen en un entorn senzill d'utilitzar. *MAPLE V* també té un llenguatge de programació que permet dissenyar procediments propis. En definitiva creiem que *MAPLE V* ha de ser un bon instrument de suport en l'estudi de les Matemàtiques en Escola Universitària Politècnica de Barcelona.

En el primer tema es fa un introducció a l'entorn de treball de *MAPLE V* i s'hi desenvolupen alguns exercicis de mostra. La intenció és que, per a cada tema, l'alumne es vagi familiaritzant amb les eines del càlcul simbòlic; no ambicionem, però, demostrar amb tota exhaustivitat tot allò que es pugui fer amb *MAPLE V* . Es tracta només que l'alumne tingui una guia d'iniciació en l'ús de l'ordinador a l'hora de fer càlculs. Més endavant, la recerca personal en punts més concrets pot ser més eficient que no pas un tractament detallat en un quadern d'aquestes característiques. I així ho esperem.

Barcelona, octubre de 1999.

Els autors.

Índex

Lliçó 1

Introducció

1.1 Descripció general de l'entorn

El disseny i la implementació del sistema MAPLE és un projecte del Symbolic Computation Group de la Universitat de Waterloo a Ontario, Canadà. Va sortir al mercat el 1988 i és comercialitzat actualment amb el nom *MAPLE V*, per "Waterloo Maple Software". Tots els comentaris i funcions que esmentarem en aquesta secció fan referència a la versió 5 i a l'entorn següent d'aquest sistema, anomenat *maple6.*

Maple és el programa de càlcul simbòlic que utilitzarem per a estudiar i resoldre alguns problemes de les diferents assignatures de Matemàtiques de les dues titulacions de l'EUPB. A més a més de ser un manipulador simbòlic, cal destacar també la seva capacitat per al càlcul numèric i per a la representacio gràfica i que disposa d'eines que fan molt versàtil la visualització de funcions, tant en el pla com a l'espai.

Per a accedir al programa cal escollir les opcions de menú "Inicio/Programas/Matemàtiques/Maple V" de l'entorn Windows.

L'entorn de treball és el propi dels programes dissenyats per a Windows: Una barra de menús desplegables, barres d'icones en funció del context en què ens trobem, finestres obertes en l'àrea de treball del programa, ajuda en línia o a través del menú i amb opcions d'hipertext, opcions d'edició de *tallar-o-copiar i enganxar* (Recordem aquesta mecànica: Una vegada sel.leccionat l'objecte -text, fórmula o gràfic-, per menú amb les opcions "Edit/Cut" o "Edit/Copy" tallem o copiem i, una vegada posicionats en el punt que ens interessa, amb "Edit/Paste" enganxem o bé amb les combinacions de tecles Ctrl-X, Ctrl-C i Ctrl-V, respectivament)...

MAPLE V consta de tres parts fonamentals: el *nucli,* que es carrega en el mateix moment de posar en marxa l'aplicació i que constitueix la part fonamental del programa ja que conté les funcions, procediments i rutines necessaris per a realitzar tots els processos bàsics (siguin simbòlics, numèrics o gràfics); la *interfície* que permet la comunicació entre l'ordinador i l'usuari i, finalment, les *llibreries* i *paquets* que constitueixen la part més important del programa ja que contenen un conjunt d'aplicacions, funcions i procediments que complementen els del nucli. El contingut de les llibreries i dels paquets es llegeix en el moment que es necessiten, de manera que el programa optimitza la disponibilitat de memòria per a l'execució dels processos.

Per a crear un document de treball -o *Worksheet* que es diu en *MAPLE V* -, només cal escollir l'opció "File/New" de la barra de menús. Si el que volem és recuperar un document ja existent, el que farem servir és l'opció "File/Open". És important recordar que, per seguretat, convé anar guardant el document de treball, amb "File/Save" o, si es vol canviar de nom el document, "File/Save As", segons ens convingui.

Quan obrim un document nou observem que a la part esquerra, i identificat amb un corxet obert ([), hi ha el que s'anomena un *grup*. Essencialment, els grups poden ser d'*execució* o de *text*. En els grups d'execució es distingeix una part *input*, que va encapçalada pel símbol > seguit de les instruccions que es volguin executar, i una part *output* en la que es visualitza el resultat de l'execució. En la configuració per defecte, a la part input la lletra és amb tipus `Courier`, de color vermell, arranada a l'esquerra i reeditable en qualsevol moment, i la part output és amb format fórmula, básicament no editable -llevat dels gràfics- i centrat respecte de l'amplada de la finestra. Els grups de text són per a introduir comentaris en el mateix document.

Per a definir un grup com a grup de text o d'execució podem fer servir la tecla F5 (o els botons Σ i T de la barra). Pel que fa a l'edició de les instruccions suposarem que el botons **x** i el que té per icona el logotip de la fulla de plàtan, *maple* en anglès, estan polsats.

La forma de funcionament del programa és interactiva, de manera que l'usuari escriu la instrucció o instruccions i tot seguit les executa i s'obté el resultat. Totes les instruccions executades queden en memòria de manera que el programa procedeix cada vegada amb l'estat actual de les variables, funcions, procediments,... Quan convingui, es poden incloure comentaris en la mateixa línia escrivint-los a continuació del símbol
sharp.

Cada instrucció necessita acabar en ":" o ";". En ambdós casos s'executa la instrucció, però si volem veure el resultat en pantalla cal acabar en ";". La instrucció s'executa en el moment de prémer la tecla Return (el cursor pot estar en qualsevol punt de la zona input del grup d'execució). Si en algun moment interessa tornar a executar algun grup només cal situar el cursor en una de les línies input del grup i prémer Return.

Per a esborrar un grup només cal sel.leccionar-lo i prémer la tecla *Del* o *Supr*. Cal tenir en compte que esborrar un grup no desfà l'acció que s'ha executat anteriorment.

1.2 L'ajuda

MAPLE V disposa d'un sistema d'ajuda molt complet, estructurat a partir d'una llista de temes genèrics cadascun dels quals permet sel.leccionar entre temes més concrets i així fins a un màxim de cinc nivells.

Els temes del primer nivell són (destaquem en negreta els que més sovint s'utilitzen i amb punts suspensius els que tenen més nivells): Example Worksheets..., **Graphics...**, Help Guide, Index..., Introduction, **Mathematics..., New User's Tour, Programming...**, System..., What's New... i Worksheet Interface...

A l'apartat **Mathematics...** hi trobem temes com Algebra..., Basic Mathematics..., Calculus..., Differential Equations..., Discrete Mathematics..., Evaluation..., Financial Functions... i altres.

Cada vegada que s'accedeix a una finestra de l'ajuda a la part superior de la finestra es visualitza l'estructura de l'ajuda i podem anar directament d'una a l'altra sel.leccionant sobre la mateixa estructura. També des de dins de cada finestra es pot navegar a altres funcions o temes relacionats amb la modalitat d'hipertext. Naturalment, a algunes de les pàgines s'hi pot accedir per més d'una via. Des d'una de les finestres podem passar a d'altres que s'hagin consultat amb els botons "←" i "→" de la barra superior de la pantalla. Tot això facilita la recerca de qualsevol informació sobre un tema o funció que ens pugui fer falta en un moment donat, sense necessitat expressa de recórrer als manuals del programa.

Les maneres d'accedir a una finestra d'ajuda són, bàsicament, tres:

- **Des d'un grup input.** Amb la funció `help` o amb l'operador ?. Per exemple, si volem informació sobre la funció `solve` (que existeix) podem fer:

  ```
  > help(solve);
  > ? solve
  ```

  i observem que hem accedit a la pàgina
  `Mathematics / Finding Roots, Fact... / solve / Overview.`

  A la finestra que se'ns ha obert trobem la sintaxi de la funció, la descripció del paràmetres que fa servir i de la mateixa funció, *links* cap altres pàgines, exemples i altres comandes relaciones que també hi podem accedir per hipertext. Per exemple, podem anar a `dsolve` que ens portarà a la pàgina
  `Mathematics / Differential Equations / dsolve / Overview`
  que també hi podíem haver accedit des de l'estructura de la part superior de la finestra (si coneixíem -o bé exploràvem- el camí).

  I des d'aquí... podem anar enrera, endavant, cap aquí, cap allà...

  Fixem-nos que si no recordàvem ben bé el nom igualment l'ajuda ens ofereix aquells termes que més s'hi assemblen. Suposem que només recordem que el nom de la funció comença per `sol`... Fem

```
> ? sol
```

i efectivament podem escollir la funció `solve` que preteníem.

- **Sel.lecció i F1.** És el sistema d'ajut més ràpid. Només cal sel.lecionar el text sobre el que es vol informació i pulsar F1. O simplement posant el cursor davant del text del qual es vol conèixer la informació i prémer Ctr+F1.

  Escriviu el terme `pi`, sel.leccioneu-lo i polseu F1. Això us portarà a l'ajuda sobre noms claus prèviament definits en el sistema i, en particular, veureu que el número π ve denotat per la variable-constant `Pi` (la P en majúscula).

- **Per recerca en el quadre de diàleg.** En el menú `Help / Topic Search` trobem un dels sistemes habituals d'ajuda en Windows. A mesura que un va escrivint el terme pel que s'interessa van apareixent aquelles entrades a l'ajuda que comencen amb les mateixes lletres... sel.leccioneu i OK. Per aquest sistema demaneu informació sobre el número e i sobre la funció `expand`.

Amb caràcter introductori us recomanem que escolliu l'opció `Help / New User's Tour` del menú. Es tracta d'un passeig per les diferents possibilitats del programa i, de manera interactiva, anireu executant i veient el funcionament bàsic del sistema.

1.3 Primers passos en *MAPLE V*

En aquesta secció presentarem, a modus d'exemple, diferents funcions bàsiques del programa corresponents a accions elementals que s'utilitzen sovint. Convé que, a mesura que es vagin presentant, les aneu executant i comproveu les possibilitats que ofereixen.

1.3.1 La inicialització

En algun moment de la sessió pot convenir "sortir de zero" una altra vegada, com si no haguéssim executat res fins al moment. Aquesta acció la fa la funció `restart`.

```
> restart;
```

1.3.2 L'assignació

MAPLE V utilitza durant tota la sessió aquelles variables que anem definint amb el contingut que els anem assignant (contingut que pot ser numèric o literal). Aquesta acció de definició i assignació d'un valor la farem amb el signe ":="

```
> a:=2/3; b:=5/7;
```

$$a := \frac{2}{3}$$

$$b := \frac{5}{7}$$

```
> a+b;
```

$$\frac{29}{21}$$

```
> pol:=x^6-3*x^4-x^5-2*x^3-3*x^2-x+1;
```

$$pol := x^6 - 3\,x^4 - x^5 - 2\,x^3 - 3\,x^2 - x + 1$$

Com podeu veure en aquesta darrera execució tots els productes cal indicar-los amb el signe * i les potències amb la notació x^a. Si en algun moment interessa utilitzar el darrer resultat obtingut en pantalla, podem utilitzar el signe % per a fer-hi referència.

1.3.3 El factoritzar i el desenvolupar

MAPLE V té dues funcions bàsiques que permeten descomposar en factors una expressió i desenvolupar les operacions indicades: `factor` i `expand`.

Si volem descomposar en els reals factors el polinomi `pol`:

```
> factor(%);
```

$$(x^2 - 3\,x + 1)\,(x + 1)^2\,(x^2 + 1)$$

I si volem desenvolupar aquest producte:

```
> expand(%);
```

$$x^6 - 3\,x^4 - x^5 - 2\,x^3 - 3\,x^2 - x + 1$$

També ens pot interessar la descomposició en els reals, amb aproximacions numèriques de les constants:

```
> factor(pol,real);
```

$$(x + 1.)^2 (x - .3819660113) (x - 2.618033989) (x^2 + 1.)$$

O la descomposició en el conjunt dels nombres complexos:

```
> factor(pol,complex);
```

$$(x + 1.)^2 (x + 1. I) (x - 1. I) (x - .3819660113) (x - 2.618033989)$$

I denota la unitat imagimaria $i = \sqrt{-1}$.

També es poden expandir altres expressions:

```
> cos(2*x)-1;
```

$$\cos(2\,x) - 1$$

```
> expand(%);
```

$$2\cos(x)^2 - 2$$

1.3.4 La simplificació

En la majoria dels processos *MAPLE V* dóna els resultats ja simplificats, però a vegades es pot demanar una simplificació complementària. La funció per a fer-ho és `simplify`.

La simplificació pot ser estrictament numèrica

```
> 4^(1/2)+2*27^(1/3);
```

$$\sqrt{4} + 2\,27^{(1/3)}$$

```
> simplify(%);
```

$$8$$

o pot involucrar d'altres funcions racionals, trigonomètriques...

```
> e := cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x);
simplify(e);
```

$$e := \cos(x)^5 + \sin(x)^4 + 2\cos(x)^2 - 2\sin(x)^2 - \cos(2\,x)$$

$$\cos(x)^5 + \cos(x)^4$$

1.3.5 La substitució

Sovint interessa substituir el valor d'una variable per una altra expressió (numèrica o literal). Aquesta acció la farem amb la funció **subs**. La funció necessita dir quines variables substituïm i en quina expressió. Naturalment el resultat de la substitució el podem assignar a una nova (o a la mateixa) variable.

Calculem el valor numèric del polinomi $p(x)$ quan x val 1:

```
> subs(x=1,pol);
```

$$-8$$

O quan s'avalua sobre l'expressió $n+m$:

```
> polnm:=subs(x=n+m, pol);
```

$$polnm := (n+m)^6 - 3\,(n+m)^4 - (n+m)^5 - 2\,(n+m)^3 - 3\,(n+m)^2 - n - m + 1$$

```
> expand(%);
```

$$\begin{aligned}&20\,n^3\,m^3 + 15\,n^2\,m^4 + 6\,n\,m^5 - 2\,n^3 - 2\,m^3 - 3\,n^4 - 3\,m^4 - n^5 - m^5 + n^6\\&\quad + m^6 + 1 + 6\,n^5\,m + 15\,n^4\,m^2 - 5\,n\,m^4 - 12\,n^3\,m - 12\,n\,m^3\\&\quad - 18\,n^2\,m^2 - 10\,n^2\,m^3 - 10\,n^3\,m^2 - 5\,n^4\,m - 3\,m^2 - 6\,n^2\,m - 6\,n\,m^2\\&\quad - 6\,n\,m - 3\,n^2 - m - n\end{aligned}$$

```
> factor(%);
```

$$(m^2 + 1 + 2\,n\,m + n^2)\,(m^2 - 3\,m + 1 + 2\,n\,m - 3\,n + n^2)\,(1 + m + n)^2$$

I si les indeterminades n i m valen 1 i 2, respectivament, obtenim

```
> subs(n=1,m=2,polnm);
```

$$160$$

que és el mateix que si haguéssim calculat el valor numèric de $p(x)$ en $x=3$:

```
> subs(x=3,pol);
```

$$160$$

1.3.6 L'avaluació

La manipulació simbòlica que el *MAPLE V* fa de totes les expressions obliga a demanar de tant en tant que avalui el valor de l'expressió o de la variable per tal de veure'n un resultat numèric. Hi ha tot un conjunt de funcions que permeten aquesta acció, totes comencen amb el terme `eval`. Tot seguit en descrivim algunes:

`eval` Fa l'avaluació d'una expressió en un punt. No cal especificar totes les variables que hi intervenen. A diferència de la funció `subs`, `eval` sap "veure" si la substitució té sentit o no.

Podem fer, per exemple, alguns dels càlculs de l'apartat anterior:

```
> eval(pol,x=1);
```

$$-8$$

```
> eval(polnm,n=1);
```

$$(1+m)^6 - 3(1+m)^4 - (1+m)^5 - 2(1+m)^3 - 3(1+m)^2 - m$$

```
> expand(%);
```

$$-4m^3 + 7m^4 + 5m^5 + m^6 - 8 - 22m^2 - 24m$$

Observem, al següent exemple, com a vegades la funció `subs` pot formular expressions que no tinguin sentit:

```
> f:=sin(x)/(cos(x)-1);
```

$$f := \frac{\sin(x)}{(\cos(x) - 1)}$$

```
> subs(x=0,f);
```

$$\frac{\sin(0)}{(\cos(0) - 1)}$$

```
> eval(f,x=0);
Error, division by zero
```

`evalc` És l'avaluador d'expressions en el conjunt dels nombres complexos. Al capítol corresponen als nombres complexos en trobareu més detalls, ara només indicarem que la unitat imaginària es denota per i majúscula I. Veiem ara com avalua l'exponencial complexa:

```
> evalc(exp(c+d*I));
```

$$e^c \cos(d) + I\, e^c \sin(d)$$

`evalf` Serveix per avaluar una expressió i donar-ne una aproximació decimal. Admet un segon paràmetre -que és opcional- per a indicar quantes xifres significatives volem que tingui el resultat. Per defecte aquest nombre és 10.

Al començament de la secció calculavem $a + b$ i el resultat venia donat en forma fraccionària irreduïble. Fem-ne ara una aproximació:

```
> evalf(a+b);
```

$$1.380952381$$

O aproximem el número π i el número e amb 5 xifres significatives:

```
> evalf(Pi,5);
```

$$3.1416$$

```
> evalf(exp(1),5);
```

$$2.7183$$

`evalm` S'utilitza per a avaluar expressions matricials, com més endavant il.lustrarem.

1.3.7 La resolució d'equacions

La funció per a buscar els zeros d'una expressió, o les solucions d'una igualtat o d'un conjunt d'igualtats, respecte del conjunt de variables que s'indiqui, és `solve`.

Calculem les arrels (reals i complexes) del polinomi $p(x)$:

```
> solve(pol);
```

$$I,\ -I,\ \frac{3}{2}+\frac{1}{2}\sqrt{5},\ \frac{3}{2}-\frac{1}{2}\sqrt{5},\ -1,\ -1$$

Resolem el sistema d'equacions $\{2x - cy = n, 3x + 2y = m\}$, respecte de les variables x i y, amb les variables c, n i m com a paràmetres:

```
> sistema:={2*x-c*y=n,3*x+2*y=m};
```

$$\mathit{sistema} := \{2\,x - c\,y = n,\ 3\,x + 2\,y = m\}$$

```
> sols:=solve(sistema,{x,y});
```

$$\mathit{sols} := \{x = \frac{c\,m + 2\,n}{4 + 3\,c},\ y = \frac{-3\,n + 2\,m}{4 + 3\,c}\}$$

Substituïnt podem comprovar les solucions obtingudes,

```
> subs(sols,sistema);
```

$$\{2\,\frac{c\,m+2\,n}{4+3\,c}-\frac{c\,(-3\,n+2\,m)}{4+3\,c}=n,\,3\,\frac{c\,m+2\,n}{4+3\,c}+2\,\frac{-3\,n+2\,m}{4+3\,c}=m\}$$

```
> simplify(%);
```

$$\{m=m,\,n=n\}$$

o trobar la solució en algun cas particular (per exemple, quan $c=\frac{1}{3}$ i $n=\frac{1}{2}$):

```
> subs(c=1/3,n=1/2,sols);
```

$$\{x=\frac{1}{15}\,m+\frac{1}{5},\,y=-\frac{3}{10}+\frac{2}{5}\,m\}$$

Si *MAPLE V* no pot trobar una expressió algebraica per a descriure les solucions les resumeix amb la funció `RootOf` actuant sobre una equació amb indeterminada $\_Z$. Aquest conjunt de solucions es pot assignar a una variable i operar, o bé se'n pot demanar una avaluació aproximada.

Suposem que volem resoldre l'equació $p(x)=x+1$ (amb $p(x)$ definit com abans):

```
> solve(pol=x+1);
```

$$0,\,-1,\,\mathrm{RootOf}(\_Z^4-2\,\_Z^3-\_Z^2-\_Z-2)$$

[1] Observem que dues de les solucions les ha pogut trobar en forma algebraica (0 i -1) però les altres les deixa indicades com les solucions que tingui el polinomi $z^4-2z^3-z^2-z-2$. Observem que si ara avaluem aquesta expressió que hem obtingut tenim una solució més:

```
> evalf(%);
```

$$0,\,-1.,\,-.8695329716$$

Però el polinomi de quart grau amb coeficient reals si té una solució real n'ha de tenir com a mínim una altra -ja que les complexes van aparellades-!!. De fet, per defecte, `RootOf` ens calcula la solució, en el terreny dels complexos, que té norma més petita. Si volem partir d'una certa aproximació de la solució només l'hem d'especificar com a paràmetre. Provem amb les aproximacions 1, 2 i 3:

```
> evalf(RootOf(z^4-2*z^3-z^2-z-2,1));
```

$$.1180468750-.9270814021\,I$$

[1] En la versió 6 de Maple apareix la funicó `RootOf` tantes vegades com el nombre d'arrels de l'argument de la funció

Hem obtingut una solució complexa,

```
> evalf(RootOf(z^4-2*z^3-z^2-z-2,2));
```

$$2.633439222$$

ara una altra de real,

```
> evalf(RootOf(z^4-2*z^3-z^2-z-2,3));
```

$$2.633439222$$

i ara la mateixa d'abans. Més endavant reprendrem aquest punt.

Hi ha una altra funció, `allvalues`, que dóna totes les arrels d'una expressió. Per exemple

```
> allvalues(RootOf(z^4-1));
```

$$1, -1, I, -I$$

mentre que

```
> evalf(RootOf(z^4-1));
```

$$-1$$

1.3.8 Els paquets i les llibreries

Els paquets i les llibreries són conjunts de funcions de les que podem disposar en el moment que les carreguem en memòria. Són una eina important en la mesura que respresenten un estalvi d'espai de memòria mentre no s'utilitzen i també perquè permeten ampliar les possibilitats del sistema amb el disseny de funcions "a mida". *MAPLE V* té paquets i llibreries específics per a diferents camps de les matemàtiques.

Per a carregar un paquet farem servir la funció `with` i per a llegir una funció de la llibreria la `readlib`.

Alguns dels paquets que farem servir són: `linalg`, que conté funcions específiques per a l'àlgebra lineal, `plots`, amb funcions bàsiques per a la visualització gràfica, `plottools`, amb eines complementàries per al tractament de gràfics, ... Si la instrucció per a llegir el paquet acaba en ";" aleshores com a ouput obtindrem totes les funcions que a partir d'aquell moment tenim disponibles; si la instrucció acaba en ":" només veiem un missatge d'avís que ens diu si hi ha funcions que utilitzin -i per tant redefineixin- un nom prèviament definit en el sistema.

```
> with(linalg);
```

[*BlockDiagonal*, *GramSchmidt*, *JordanBlock*, *LUdecomp*, *QRdecomp*, *Wronskian*, *addcol*, *addrow*, *adj*, *adjoint*, *angle*, *augment*, *backsub*, *band*, *basis*, *bezout*, *blockmatrix*, *charmat*, *charpoly*, *cholesky*, *col*, *coldim*, *colspace*, *colspan*, *companion*, *concat*, *cond*, *copyinto*, *crossprod*, *curl*, *definite*, *delcols*, *delrows*, *det*, *diag*, *diverge*, *dotprod*, *eigenvals*, *eigenvalues*, *eigenvectors*, *eigenvects*, *entermatrix*, *equal*, *exponential*, *extend*, *ffgausselim*, *fibonacci*, *forwardsub*, *frobenius*, *gausselim*, *gaussjord*, *geneqns*, *genmatrix*, *grad*, *hadamard*, *hermite*, *hessian*, *hilbert*, *htranspose*, *ihermite*, *indexfunc*, *innerprod*, *intbasis*, *inverse*, *ismith*, *issimilar*, *iszero*, *jacobian*, *jordan*, *kernel*, *laplacian*, *leastsqrs*, *linsolve*, *matadd*, *matrix*, *minor*, *minpoly*, *mulcol*, *mulrow*, *multiply*, *norm*, *normalize*, *nullspace*, *orthog*, *permanent*, *pivot*, *potential*, *randmatrix*, *randvector*, *rank*, *ratform*, *row*, *rowdim*, *rowspace*, *rowspan*, *rref*, *scalarmul*, *singularvals*, *smith*, *stackmatrix*, *submatrix*, *subvector*, *sumbasis*, *swapcol*, *swaprow*, *sylvester*, *toeplitz*, *trace*, *transpose*, *vandermonde*, *vecpotent*, *vectdim*, *vector*, *wronskian*]

Alguna de les funcions que podem carregar directament de la llibreria són: `extrema`[2], per al càlcul dels extrems d'una funció de vàries variables, `polar`, per a la representació polar de nombres complexos, ... Si la instrucció per a llegir la funció acaba en ";" aleshores com a ouput veurem els paràmetres amb què treballa. Si el nom de la funció ha estat utilitzat prèviament aleshores haurem de desassignar abans la funció amb `unassign` o escriure el nom de la funció entre cometes (cometes simples, no dobles).

Suposem que a la variable `extrema` hi tenim el valor 1 [3]:

```
> extrema:=1;
```

$$extrema := 1$$

Quan intentem llegir la funció veiem que ens dóna error (de fet estem fent `readlib(1)`!!):

```
> readlib(extrema);

Error, wrong number (or type) of parameters in function readlib
```

Correctament, hauríem de fer:

```
> unassign('extrema'):  readlib('extrema');
```

[2]En la versió 6 no cal carregar aquesta llibreria ja que la funció està incorporada al propi programa

[3]En Maple 6, quan assignem a `extrema` el valor 1, el programa ja ens avisa de que estem fent una assignació incorrecta

proc(*fcn*, *cnstrnts*, *vars*, *candidates*) ... **end**

1.3.9 Operant amb vectors

En una primera aproximació, la definició d'un vector es pot fer a partir de la utilització d'una llista que ve indicada per corxets ([ ]) i a les components hi podem fer referència mitjançant la posició que ocupen. La funció `type` ens permet verificar si l'objecte creat és un vector o no.

Anem a definir el "vector" v com a llista:

```
> v:=[1,-1,2];
```

$$v := [1, -1, 2]$$

Podem comprovar que v no és de fet un vector,

```
> type(v,vector);
```

false

tot i que podem fer referència a les components de la llista:

```
> v[1];v[3];
```

1

2

Si volem definir pròpiament les variables com a vectors ens cal utilitzar la funció `array` del nucli de *MAPLE V* o la `vector` del paquet `linalg` d'àlgebra lineal. En el cas de la funció `array` és opcional indicar la seqüència d'índexos -que per defecte serà l'1 i següents- i si fem servir la funció `vector` és opcional indicar la dimensió del vector -que per defecte serà la longitud de la llista proposada-.

```
> v:=array([1,-1,2]);
```

$$v := [1, -1, 2]$$

```
> type(v,vector);
```

true

```
> with(linalg):  w:=vector([2,1,3]);
Warning, new definition for norm
Warning, new definition for trace
```

$$w := [2,\ 1,\ 3]$$

Una vegada tenim definits vectors podem fer combinacions lineals, productes escalars, productes vectorials, càlcul de normes

```
> 3*v-5*w;
```

$$3\,v - 5\,w$$

```
> evalm(%);
```

$$[-7,\ -8,\ -9]$$

```
> dotprod(v,w);
```

$$7$$

```
> u:=crossprod(v,w);
```

$$u := [-5,\ 1,\ 3]$$

```
> norm(w,2);
```

$$\sqrt{14}$$

Podem comprovar fàcilment que el producte vectorial u, ha quedat ortogonal als vectors v i w:

```
> dotprod(u,v);dotprod(u,w);
```

$$0$$

$$0$$

1.3.10 Operant amb matrius

La definició d'una matriu és similar a la dels vectors si fem servir la funció `array`. El paquet `linalg` té una funció específica per a tal fi: `matrix`. En aquesta funció és necessari especificar el nombre de files i de columnes que ha de tenir la matriu i el contingut de la matriu per files.

Les tres següents accions són equivalents:

```
> A:=array([[1,-2,4],[1,2,1]]):
A:=array(1..2,1..3,[[1,-2,4],[1,2,1]]):
A:=matrix(2,3,[1,-2,4,1,2,1]);
```

$$A := \begin{bmatrix} 1 & -2 & 4 \\ 1 & 2 & 1 \end{bmatrix}$$

```
> type(A,matrix);
```

$$true$$

Definim una altra matriu i fem les operacions matricials bàsiques:

```
> B:=array([[-2,c],[1,-1],[-3,c]]);
```

$$B := \begin{bmatrix} -2 & c \\ 1 & -1 \\ -3 & c \end{bmatrix}$$

Combinació lineal i transposició:

```
> A + transpose(B);
```

$$A + \begin{bmatrix} -2 & 1 & -3 \\ c & -1 & c \end{bmatrix}$$

```
> evalm(%);
```

$$\begin{bmatrix} -1 & -1 & 1 \\ 1+c & 1 & 1+c \end{bmatrix}$$

El producte matricial (que ve denotat pel signe &*), en el cas de la matriu A operant sobre el vector v, o el producte de les matrius A i B:

```
> evalm(A &* v);
```

$$[11, 1]$$

```
> C:=A &* B;
```

$$C := A \,\&{*}\, B$$

```
> evalm(%);
```

$$\begin{bmatrix} -16 & 5c+2 \\ -3 & 2c-2 \end{bmatrix}$$

De forma equivalent opera la funció `multiply`

```
> multiply(A,B);
```

$$\begin{bmatrix} -16 & 5c+2 \\ -3 & 2c-2 \end{bmatrix}$$

Podem calcular determinants, amb la comanda **det**, inverses amb **inverse**, traces amb **trace** ...

```
> det(C);
```

$$-17\,c+38$$

```
> inverse(C);
```

$$\begin{bmatrix} -2\,\dfrac{-1+c}{17\,c-38} & \dfrac{5\,c+2}{17\,c-38} \\ -3\,\dfrac{1}{17\,c-38} & 16\,\dfrac{1}{17\,c-38} \end{bmatrix}$$

```
> trace(C);
```

$$-18+2\,c$$

1.3.11 Les funcions i els procediments

A vegades ens convindrà definir noves funcions que a partir d'uns paràmetres inicials calculin un nou valor (escalar, vectorial, de text, lògic,...). Aquestes funcions les definirem a partir de l'operador "$->$" tal com il.lustrem tot seguit.

Anem a definir la funció $f(x)=\frac{x^3-x+1}{x-1}$:

```
> f:=x->(x^3-x+1)/(x-1);
```

$$f:=x\to\frac{x^3-x+1}{x-1}$$

A diferència del polinomi p, definit al principi de la secció per assignació directa i que només podem avaluar amb **subs** o **eval**, la funció $f(x)$ és efectivament una funció i podem calcular directament imatges:

```
> f(3);
```

$$\frac{25}{2}$$

```
> f(x-y);
```

$$\frac{(x-y)^3-x+y+1}{x-y-1}$$

Anem a definir ara una funció, que anomenarem *sumapot*, que calculi la suma de les potències n-èssimes de dos valors que li passem. Els dos primers paràmetres seran els valors i el tercer l'exponent:

```
> sumapot:=(x,y,e)->x^e+y^e;
```

$$sumapot := (x, y, e) \to x^e + y^e$$

Comprovem el seu funcionament:

```
> sumapot(1,-1,2);
```

$$2$$

```
> sumapot(n-m,k,p);
```

$$(n-m)^p + k^p$$

Un conjunt d'accions a partir d'uns paràmetres inicials també es poden englobar en forma del que s'anomena un *procediment* i que es defineixen a partir de l'estructura `proc(...)...end`. Els procediments a diferència de les funcions poden retornar a l'usuari més d'un valor.

En el moment de la definició, a l'epígraf `local` s'identifiquen les variables que només s'utilitzaran temporalment en el procediment. A l'epígraf `global` s'identifiquen les variables el valor de les quals serà llegit i actualitzat en tot l'entorn. Si no se'n fa definició explícita, totes les variables que apareixen en el procediment són considerades globals excepte les variables utilitzades com a comptador en una sentència tipus `for` o les variables a les quals se'ls fa una assignació.

En acabar, la funció RETURN indica quins valors es presenten a l'output del grup d'execució. Si no se'n fa ús, el procediment mostra el valor de l'última execució.

Per a canviar de línia durant l'edició del procediment sense que s'executi el grup podem utilitzar la combinació Shift+Return.

Construïm ara un procediment, que en direm *tot*, que, donats dos vectors ens en calcularà el producte escalar, el producte vectorial, les normes dels dos vectors i la del producte vectorial, i l'angle (en graus sexagesimals) que formen els dos vectors. Observem que la variable *esc* serà local i que la variable *vec* global:

```
> tot:=proc(v1,v2) local esc,nv1,nv2,nv3,alpha; global vec;
esc:=dotprod(v1,v2); vec:=crossprod(v1,v2);
nv1:=norm(v1,2); nv2:=norm(v2,2); nv3:=norm(vec,2);
alpha:=arccos(esc/(nv1*nv2));
RETURN(esc,evalm(vec),nv1,nv2,nv3,evalf(alpha*180/Pi)); end;
```

$$
\begin{array}{l}
tot := \textbf{proc}(v1,\ v2) \\
\quad \textbf{local}\ esc,\ nv1,\ nv2,\ nv3,\ \alpha; \\
\quad \textbf{global}\ vec; \\
\qquad esc := \mathrm{dotprod}(v1,\ v2)\,; \\
\qquad vec := \mathrm{crossprod}(v1,\ v2)\,; \\
\qquad nv1 := \mathrm{norm}(v1,\ 2)\,; \\
\qquad nv2 := \mathrm{norm}(v2,\ 2)\,; \\
\qquad nv3 := \mathrm{norm}(vec,\ 2)\,; \\
\qquad \alpha := \arccos(esc\,/(nv1 \times nv2))\,; \\
\qquad \mathrm{RETURN}(esc,\ \mathrm{evalm}(vec),\ nv1,\ nv2,\ nv3,\ \mathrm{evalf}(180 \times \alpha\,/\pi)) \\
\quad \textbf{end}
\end{array}
$$

Apliquem la funció sobre els vectors $(1,2,3)$ i $(-1,2,1)$:

```
> tot(vector([1,2,3]),vector([-1,2,1]));
```

$$6,\ [-4,\ -4,\ 4],\ \sqrt{14},\ \sqrt{6},\ 4\sqrt{3},\ 49.10660534$$

Una vegada executat el procediment comprovem que la variable *esc* no existeix, mentre que s'ha creat la variable *vec* i té el valor previst:

```
> esc;
```

$$esc$$

```
> evalm(vec);
```

$$[-4,\ -4,\ 4]$$

També en els procediments s'utilitzen estructures de programació habituals. Sense voler entrar en detall, comprovem com el procediment *successio* permet obtenir una col.lecció de termes (des del subíndex *inici* fins al subíndex *final*, de la successió que té de terme general *tergen*:

```
> successio:=proc(inici,final,tergen) local i; for i from inici to final
do print(i,tergen(i)) od; end;
```

$$
\begin{array}{l}
successio := \textbf{proc}(inici,\ final,\ tergen) \\
\quad \textbf{local}\ i; \\
\qquad \textbf{for}\ i\ \textbf{from}\ inici\ \textbf{to}\ final\ \textbf{do}\ \mathrm{print}(i,\ tergen(i))\ \textbf{od} \\
\quad \textbf{end}
\end{array}
$$

Apliquem-ho a la nostra funció $f(x)$:

```
> successio(2,7,f);
```

$$2,\ 7$$

$$3, \frac{25}{2}$$

$$4, \frac{61}{3}$$

$$5, \frac{121}{4}$$

$$6, \frac{211}{5}$$

$$7, \frac{337}{6}$$

1.3.12 Els gràfics

Tot i que el mateix nucli del *MAPLE V* ja proporciona dues funcions: `plot`, per a gràfics en el pla i `plot3d`, per als gràfics tridimiensionals, la gran majoria de funcions per a la visualització de gràfics estan disponibles després de carregar el paquet `plots` A vegades ens pot fer falta també el `plottools`.

Observeu, pel mateix nom, les possibilitats gràfiques disponibles. Les múltiples opcions disponibles per aquestes funcions (com ara colors, tipus de linies, text, eixos, caixes, ...) es poden consultar a l'Ajuda del programa. Aquí només n'il.lustrarem les possibilitats bàsiques.

```
> with(plots)
```

[*animate*, *animate3d*, *animatecurve*, *changecoords*, *complexplot*, *complexplot3d*, *conformal*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *cylinderplot*, *densityplot*, *display*, *display3d*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *odeplot*, *pareto*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *replot*, *rootlocus*, *semilogplot*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *sphereplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

```
> help(plot);
```

Tornant a la resolució de l'equació polinòmica $p(x) = x + 1$ de la subsecció 1.3.7, podem ara representar la funció $p(x) - x - 1$ i mirar en quins moments talla l'eix de les x's:

```
> plot(pol-x-1, x=-5..5);
```



Observeu que aquesta visualització ens mostra el comportament de la funció lluny de l'origen però no ens ajuda gaire a veure les solucions de l'equació. També podeu veure que la sel.lecció automàtica que hi hagut del rang en l'eix de les y's fa que aquest tingui una escala "molt gran" respecte el de les x's. De moment podem fer una aproximació visualitzant només la funció a l'interval [-2,3]:

```
> plot(pol-x-1,x=-2..3);
```



El resultat és força més clar, tot i que l'existència de les solucions $x = 0$ i $x = -1$ ens fa difícil veure la solució que busquem. Reduïm la funció i representem ara el polinomi

$\dfrac{p(x) - x - 1}{x(x+1)}$, en el mateix interval [-2,3]:

```
> plot((pol-x-1)/(x*(x+1)),x=-2..3);
```



Podeu veure que la solució trobada es correspon amb la que havíem calculat a l'apartat 1.3.7.

La funció `plot` també permet la representació simultània d'una llista de funcions. Les opcions que vulguem utilitzar també es poden definir en una llista. Veiem-ho amb el següent exemple i representem les funcions $y = \sin x$ i $y = 2\sin(3x - 6) + 3$:

```
> plot([sin(x),2*sin(3*x-6)+3], x=-Pi..Pi,color=[red,blue],
style=[point,line]);
```



El mateix resultat gràfic el podem obtenir a partir de dues variables que guardin cadascun dels gràfics, per separat, i fent servir la funció `display` per a mostrar-los alhora:

```
> sinugraf:=plot(sin(x),x=-Pi..Pi,color=red,style=point):
sinugrafmod:=plot(2*sin(3*x-6)+3,x=-Pi..Pi,color=blue,style=line):
display(sinugraf,sinugrafmod);
```



Les possibilitats gràfiques per a representació de corbes i superfícies a l'espai són moltes i en el capítol que dediquem a aquest tema ja hi entrem en profunditat. Només amb caràcter de *demo*, podem veure, ràpidament, com *MAPLE V* és capaç de representar animacions gràfiques. L'exemple següent correspon al quadrat $[0, \pi] \times [0, \pi]$ sortint d'una posició plana (alçada 0) i anant-se modulant en introduir cosinusoides en la direcció de les x's i sinusoides en la direcció de les y's, a mesura que avança el paràmetre (de 0 a 2).

```
>  animate3d(cos(t*x)*sin(t*y),x=0..Pi, y=0..Pi,t=0..2)
```



Una vegada executada la instrucció, per a visualitzar l'animació cal fer un click sobre el gràfic i veurem que en aquell moment apareix una barra de botons; només cal aleshores prémer el *play* que correspon a la icona ▷:



1.3.13 Una mica de Càlcul

Estudiem, per començar, el límit en el 0 de la funció $f(x) = xe^{\frac{-1}{x}}$. La funció de *MAPLE V* que farem servir és `limit`:

```
> limit(x*exp(-1/x),x=0);
```

$$undefined$$

Veiem doncs que el límit no està definit. Calculem ara els límits laterals:

```
> limit(x*exp(-1/x),x=0,left);
```

$$-\infty$$

```
> limit(x*exp(-1/x),x=0,right);
```

$$0$$

Considerem ara la funció polinòmica $f(x) = x^6 + 5x^5 - 22x^3 + x^2 + 33x - 18$. Anem a calcular els seus intervals de creixement/decreixement i de concavitat/convexitat així com els punts notables i altres elements de la gràfica de la funció.

Per al càlcul de la funció derivada necessitarem la funció de *MAPLE V* `diff`. La funció serveix per a calcular les derivades respecte de qualssevol de les variables i de l'ordre que s'indiqui.

Definim la funció f i les seves primeres tres derivades (noteu que les funcions *f1, f2* i *f3* calculen per a cada punt x la corresponent derivada en el punt:

```
> f:=x->x^6+5*x^5-22*x^3+x^2+33*x-18;
f1:=x->eval(diff(f(t),t),t=x);f1(x);
f2:=x->eval(diff(f(t),t$2),t=x);f2(x);
f3:=x->eval(diff(f(t),t$3),t=x);f3(x);
```

$$f := x \rightarrow x^6 + 5\,x^5 - 22\,x^3 + x^2 + 33\,x - 18$$

$$f1 := x \rightarrow \left(\frac{\partial}{\partial t}\mathrm{f}(t)\right)\Big|_{t\,=\,x}$$

$$6\,x^5 + 25\,x^4 - 66\,x^2 + 2\,x + 33$$

$$f2 := x \rightarrow \left(\frac{\partial^2}{\partial t^2}\mathrm{f}(t)\right)\Big|_{t\,=\,x}$$

$$30\,x^4 + 100\,x^3 - 132\,x + 2$$

$$f3 := x \to \left(\frac{\partial^3}{\partial t^3} \mathrm{f}(t)\right)\Big|_{t\,=\,x}$$

$$120\,x^3 + 300\,x^2 - 132$$

Calculem el punts d'intersecció amb l'eix de les x's:

```
> solve(f(x));
```

$$-2,\ -3,\ -3,\ 1,\ 1,\ 1$$

Calculem els punt que anul.len la primera derivada, que seran els possibles punts de màxim o mínim:

```
> pmaxmin:=solve(f1(x));
```

$$pmaxmin := -3,\ -\frac{19}{12} + \frac{1}{12}\sqrt{97},\ -\frac{19}{12} - \frac{1}{12}\sqrt{97},\ 1,\ 1$$

```
> pmaxmin:=evalf(%);
```

$$pmaxmin := -3.,\ -.7625951829,\ -2.404071483,\ 1.,\ 1.$$

Substituïm aquests punts a la derivada segona

```
> f2(-3);f2(pmaxmin[3]);f2(pmaxmin[2]);f2(1);
```

$$128$$

$$-68.0107212$$

$$68.45979487$$

$$0$$

i veiem que en $x = -3$ i en $x = \dfrac{-19+\sqrt{97}}{12}$ hi ha un mínim i que en $x = \dfrac{-19-\sqrt{97}}{12}$ hi ha un màxim. El punt $x = 1$ apareix com a candidat a ser inflexió vist que anul.la la segona derivada. Els intervals de creixement/decreixement els podem confeccionar a partir dels punts trobats.

Anem a refer el mateix amb les derivades segona i tercera per als punt d'inflexió i per als intervals de concavitat/convexitat:

```
> pinflex:=solve(f2(x));
```

$$pinflex := 1,\ \frac{2}{45}\%1^{(1/3)} + \frac{130}{9}\frac{1}{\%1^{(1/3)}} - \frac{13}{9},$$
$$-\frac{1}{45}\%1^{(1/3)} - \frac{65}{9}\frac{1}{\%1^{(1/3)}} - \frac{13}{9} + \frac{1}{2}I\sqrt{3}\,(\frac{2}{45}\%1^{(1/3)} - \frac{130}{9}\frac{1}{\%1^{(1/3)}}),$$
$$-\frac{1}{45}\%1^{(1/3)} - \frac{65}{9}\frac{1}{\%1^{(1/3)}} - \frac{13}{9} - \frac{1}{2}I\sqrt{3}\,(\frac{2}{45}\%1^{(1/3)} - \frac{130}{9}\frac{1}{\%1^{(1/3)}})$$
$$\%1 := 1700 + 675\,I\sqrt{69}$$

```
> evalf(%);
```

$$1.,\ .015154164 + .1\,10^{-9}\,I,\ -2.747048125 + .1\,10^{-9}\,I,$$
$$-1.601439372 - .1\,10^{-9}\,I$$

Atenció: Hem obtingut tres solucions complexes i una de real en un polinomi de grau 4 a coeficients reals!! Això ens diu que les aproximacions del `solve` i de l'`evalf` no han estat prou precises. Noteu, però, que els nombres complexos obtinguts tenen una part imaginària quasi zero (i si augmentem la precisió de la funció `evalf` encara més) cosa que ens indica que potser són solucions reals. El que farem serà fer la resolució directament amb aproximació per coma o punt flotant. Això ens ho farà la funció `fsolve`:

```
> pinflex:=fsolve(f2(x));
```

$$pinflex := -2.747048125,\ -1.601439372,\ .01515416360,\ 1.$$

Observeu que les aproximacions trobades són coherents amb les anteriors. Avaluem ara la tercera derivada en aquests punts:

```
> f3(pinflex[1]);f3(pinflex[2]);f3(pinflex[3]);f3(1);
```

$$-355.715124$$

$$144.5346996$$

$$-131.9306878$$

$$288$$

Veiem que efectivament els quatre punts corresponen a inflexions, canviant la funció f de convexa a còncava en el primer i el tercer i de còncava a convexa en el segon i el quart. (Alternativament, també podíem haver avaluat la derivada segona a l'entorn o entre les inflexions.)

Podem fer ara la gràfica conjunta de la funció f i les seves dues primeres derivades. Comproveu com el signe de la derivada primera control.la el creixement/decreixement de la funció, i com el de la segona ho fa per a la concavitat/convexitat:

```
> plot([f(x),f1(x),f2(x)],x=-4..4,y=-75..75);
```



Per a acabar aquesta subsecció, calculem ara algunes integrals. La funció que farem servir és `int` (Per a la presentació formal de les integrals es fa servir la funció `Int`). Calculem, per exemple

$$\int \frac{1+\sin x}{\cos x} dx$$

```
> int((1+sin(x))/cos(x),x);
```

$$\ln(\sec(x)+\tan(x))-\ln(\cos(x))$$

O la mateixa integral però entre 0 i $\frac{\pi}{4}$

```
> int((1+sin(x))/cos(x),x=0..Pi/4);
```

$$\frac{1}{2}\ln(2)+\ln(\sqrt{2}+1)$$

I com aplicació podem calcular l'àrea del recinte, en el semipla $x > 0$, limitat -simultàniament- per la funció f, les seves dues primeres derivades *f1, f2* i l'eix de les y's (vegeu la figura anterior).

Trobem primer el punt intersecció (de fet, tots) entre la funció i la derivada segona:

```
> inter:=fsolve(f(x)=f2(x));
```

$$\mathit{inter} := -6.852908091, -2.753775251, -1.676227563, .1225211642, 1., 5.160389741$$

Observeu que el punt que busquem és el quart dels que hem obtingut. I calculem ara les dues integrals en qüestió:

```
> int(f1(x)-f2(x),x=0..inter[4])+int(f1(x)-f(x),x=inter[4]..1);
```

$$22.47207369$$

1.4 Exercicis

Aquesta és una relació d'exercicis molt fàcils per tal d'introduir-vos en algunes de les comandes que s'han vist a la secció anterior. Tot i que alguns són molt senzills de fer en el paper, us recomanem que els resolgueu amb el *MAPLE V* .

1. Factoritzeu els polinomis següents:

a) $x^4 + 2x^3 - 3x^2 - 4x + 4$,

b) $x^4 - 6x^3 - 11x^2 + 96x - 80$.

2. Resoleu les equacions següents:

a) $6x^5 - 37x^4 - 31x^3 - 30x^2 + 13x + 7 = 0$,

b) $4x^4 - 37x^2 + 9 = 0$.

3. Calculeu:

$$\sqrt[5]{1+i}$$

4. Donats els punts $A(-3,1)$, $B(-2,-5)$ i $C(4,-1)$, calculeu:

a) l'equació de la recta $\mathbf{r}$ que passa pels punts A, B,

b) el punt simètric del C respecte de la recta $\mathbf{r}$,

c) l'àrea del triangle de vèrtexs A, B i C.

5. Determineu:

a) el valor d'a de manera que els punts $A(a,3); B(-5,1); C(1,2)$ estiguin alineats.

b) el vèrtex C d'un triangle si coneixeu els vèrtexs $A(4,3), B(10,-5)$ i el baricentre G(5, -1).

6. Resoleu els següents sistemes d'equacions:

$$\text{a)} \left\{ \begin{array}{rcl} 4x + y - 2z & = & -5 \\ 3x + 2y - 5z & = & -4 \\ x - y + 3z & = & -1 \end{array} \right. \quad \text{b)} \left\{ \begin{array}{rcr} x + 2y + z + 2t & = & 0 \\ -x - 2y + 3z + 2t & = & 1 \\ 8x + 16y - 4z + 4t & = & -3 \end{array} \right. .$$

7. Estudieu i representeu gràficament les funcions:

a) $y = a \sin bx$, per $a = 2$, $b = 1$ i per $a = 1$, $b = 2$;

b) $y = a^x$ i $y = 2^{-ax}$, per $a = 2$ i 4;

c) $f(x) = \dfrac{\ln x}{x}$;

d) $f(x) = \dfrac{4x^k - 3}{x^3 + 5}$, per a $k = 2, 3, 4$.

8. Sigui la funció $f(x) = \cos^2 x$.

a) Calculeu una primitiva de $f(x)$.

b) Trobeu l'àrea del recinte limitat per la funció i els eixos, entre 0 i $\frac{\pi}{2}$.

c) Calculeu el volum de la figura de revolució que obtenim en fer girar el recinte de l'apartat anterior al voltant de l'eix de les x's.

Lliçó 2

Àlgebra Lineal

Amb *MAPLE V* es poden realitzar moltes de les operacions habituals de l'Àlgebra Lineal, com ara multiplicar matrius, resoldre sistemes d'equacions, diagonalitzar endomorfismes, etc. Al capítol d'Àlgebra Lineal del **Help** podeu trobar una relació completa de totes aquestes operacions, il.lustrades amb exemples.

Moltes de les utilitats d'Àlgebra Lineal només funcionen carregant prèviament el paquet `linalg`:

```
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

En aquest capítol treballarem doncs amb el paquet `linalg`.

A la Introducció s'han vist ja les diferents maneres de definir vectors i matrius i les principals operacions de l'Àlgebra Lineal, recordem algunes de les sentències de *MAPLE V* que més s'utilitzaran aquí:

Definició d'un vector	u:=`vector([ , ,...,])`
Norma d'un vector	`norm(u,2)`
Producte escalar de vectors	`dotprod(u,v)`
Producte vectorial	`crossprod(u,v)`
Definició d'una matriu	A:=`matrix(m,n,[ , ,...,])`
Matriu identitat $n \times n$	`array(identity, 1..n,1..n)`
Suma de matrius	A+B
Producte per escalars	`evalm(k*A)`
Producte de matrius	`evalm(A &* B)` o `multiply(A,B)`
Determinant	`det(A)`
Trasposta	`transpose(A)`
Traça	`trace(A)`
Inversa	`inverse(A)`
Gauss	`gausselim(A)`
Gauss–Jordan	`gaussjord(A)`
Resolució de sistemes lineals	`linsolve(A,b)`
Nucli d'aplicacions lineals	`kernel(A)`
Polinomi característic	`charpoly(A,x)`
Valors propis	`eigenvalues(A)`
Vectors propis	`eigenvectors(A)`

A continuació es resolen alguns problemes d'Àlgebra Lineal, fent servir *MAPLE V*.

2.1 Resolució de sistemes d'equacions

Exemple 1 *Resoleu els següents sistemes:*

$$\left\{\begin{array}{ll} x-y+2z & =3, \\ 2x+y+3z & =5, \\ x-2y+3z & =4. \end{array}\right. ;\left\{\begin{array}{ll} x-y+2z & =3, \\ 2x+y+3z & =5, \\ 3x+5z & =8. \end{array}\right. ;\left\{\begin{array}{ll} x-y+2z & =3, \\ 2x+y+3z & =5, \\ -3y+z & =4. \end{array}\right.$$

Els sistemes donats es poden resoldre directament amb la instrucció solve:

```
> solve({x-y+2*z=3,2*x+y+3*z=5,x-2*y+3*z=4},{x,y,z});
```

$$\{y=0,\ z=1,\ x=1\}$$

```
> solve({x-y+2*z=3,2*x+y+3*z=5,3*x+5*z=8},{x ,y,z});
```

$$\{z=3y+1,\ x=-5y+1,\ y=y\}$$

```
> solve({x-y+2*z=3,2*x+y+3*z=5,-3*y+z=4},{x, y,z});
```

Fem servir el llenguatge de matrius per a resoldre sistemes d'equacions lineals. Definim la matriu de coeficients del sistema i el vector dels termes independents:

```
> with(linalg):
> A1:=matrix(3,3,[1,-1,2,2,1,3,1,-2,3]);
```

$$A1 := \begin{bmatrix} 1 & -1 & 2 \\ 2 & 1 & 3 \\ 1 & -2 & 3 \end{bmatrix}$$

```
> b1:=vector([3,5,4]);
```

$$b1 := [3, 5, 4]$$

Podem resoldre el sistema directament amb la instrucció `linsolve`:

```
> linsolve(A1,b1);
```

$$[1, 0, 1]$$

També podem construir la matriu ampliada amb la instrucció **concat** i aplicar Gauss amb **gausselim**. Observem però, que el resultat és poc pràctic perquè encara cal fer alguns càlculs per a obtenir la solució:

```
> concat(A1,b1);
```

$$\begin{bmatrix} 1 & -1 & 2 & 3 \\ 2 & 1 & 3 & 5 \\ 1 & -2 & 3 & 4 \end{bmatrix}$$

```
> gausselim(%);
```

$$\begin{bmatrix} 1 & -1 & 2 & 3 \\ 0 & 3 & -1 & -1 \\ 0 & 0 & \frac{2}{3} & \frac{2}{3} \end{bmatrix}$$

mentre que si fem servir la instrucció **gaussjord**, la solució és més immediata.

```
> gaussjord(concat(A1,b1));
```

$$\begin{bmatrix} 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 1 \end{bmatrix}$$

Fem el mateix pel segon sistema.

```
> A2:=matrix(3,3,[1,-1,2,2,1,3,3,0,5]);
```

$$A2 := \begin{bmatrix} 1 & -1 & 2 \\ 2 & 1 & 3 \\ 3 & 0 & 5 \end{bmatrix}$$

```
> b2:=vector([3,5,8]);
```

$$b2 := [3, 5, 8]$$

En aquest cas ja hem vist que el sistema és compatible indeterminat. El paràmetre $\_t_1$ és el que representa el grau de llibertat del sistema.

```
> linsolve(A2,b2);
```

$$[1 - 5\_t_1,\ \_t_1,\ 3\_t_1 + 1]$$

Fent servir les comandes **concat** i **gaussjord** obtenim

```
> gaussjord(concat(A2,b2));
```

$$\begin{bmatrix} 1 & 0 & 5/3 & 8/3 \\ 0 & 1 & -1/3 & -1/3 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

En el tercer sistema no s'ha obtingut cap solució, perquè és incompatible.

```
> A3:=matrix(3,3,[1,-1,2,2,1,3,0,-3,1]);
```

$$A3 := \begin{bmatrix} 1 & -1 & 2 \\ 2 & 1 & 3 \\ 0 & -3 & 1 \end{bmatrix}$$

```
> b3:=vector([3,5,4]);
```

$$b3 := [3, 5, 4]$$

```
> linsolve(A3,b3);
```

La incompatibilitat del sistema es pot observar també

```
> concat(A3,b3);
```

$$\begin{bmatrix} 1 & -1 & 2 & 3 \\ 2 & 1 & 3 & 5 \\ 0 & -3 & 1 & 4 \end{bmatrix}$$

```
> gausselim(%);
```

$$\begin{bmatrix} 1 & -1 & 2 & 3 \\ 0 & 3 & -1 & -1 \\ 0 & 0 & 0 & 3 \end{bmatrix}$$

Exemple 2 *Considereu el següent sistema amb els paràmetres k i e.*

$$\left\{ \begin{aligned} kx + 6y + z - 3t &= 2 \\ 2x - 2y - 5z + 9t &= 0 \\ x + ky + 2z - 5t &= e \\ 2x + 3y - z + 2t &= e \end{aligned} \right.$$

Discutiu i busqueu les solucions del sistema segons els valors dels paràmetres.

Introduïm la matriu de coeficients del sistema:

```
> with(linalg):
> A:=matrix(4,4,[k,6,1,-3,2,-2,-5,9,1,k,2,-5,2, 3,-1,2]);
```

$$A := \begin{bmatrix} k & 6 & 1 & -3 \\ 2 & -2 & -5 & 9 \\ 1 & k & 2 & -5 \\ 2 & 3 & -1 & 2 \end{bmatrix}$$

Busquem el determinant:

```
> det(A);
```

$$33\,k + k^2 - 108$$

i per quins valors de k s'anul.la:

```
> solve(%,k);
```

$$-36,\ 3$$

Per tot valor de $k \neq 3, -36$ el sistema és compatible i determinat, té una única solució:

```
> b:=vector([2,0,e,e]);
```

$$b := [2,\ 0,\ e,\ e]$$

```
> linsolve(A,b);
```

$$\left[2\,\frac{23 - 35\,e + 3\,k\,e + k}{33\,k + k^2 - 108},\ 2\,\frac{-13 + e + 4\,k\,e}{33\,k + k^2 - 108},\ \frac{162 - 54\,e - 35\,k\,e - 28\,k + 9\,k^2\,e}{33\,k + k^2 - 108},\right.$$
$$\left.\frac{-16\,k - 19\,k\,e + 5\,k^2\,e + 74 - 14\,e}{33\,k + k^2 - 108}\right]$$

Per $k = 3$ s'obté la matriu $A1$.

```
> A1:=subs(k=3, evalm(A));
```

$$A1 := \begin{bmatrix} 3 & 6 & 1 & -3 \\ 2 & -2 & -5 & 9 \\ 1 & 3 & 2 & -5 \\ 2 & 3 & -1 & 2 \end{bmatrix}$$

la matriu ampliada

```
> A1b:=concat(A1,b)
```

$$A1b := \begin{bmatrix} 3 & 6 & 1 & -3 & 2 \\ 2 & -2 & -5 & 9 & 0 \\ 1 & 3 & 2 & -5 & e \\ 2 & 3 & -1 & 2 & e \end{bmatrix}$$

Si fem

```
> gausselim(A1b);
```

$$\begin{bmatrix} 3 & 6 & 1 & -3 & 2 \\ 0 & -6 & \frac{-17}{3} & 11 & \frac{-4}{3} \\ 0 & 0 & \frac{13}{18} & \frac{-13}{6} & e - \frac{8}{9} \\ 0 & 0 & 0 & 0 & 2e-2 \end{bmatrix}$$

que per $e \neq 1$ dóna un sistema incompatible, mentre que si $e = 1$ el sistema és compatible indeterminat amb un grau de llibertat:

```
> b1:=subs(e=1,evalm(b));
```

$$b1 := [2, 0, 1, 1]$$

```
> linsolve(A1,b1);
```

$$\left[\frac{6}{13} + 2\_t_1, \frac{1}{13} - \_t_1, \frac{2}{13} + 3\_t_1, \_t_1\right]$$

Fem el mateix per $k = -36$.

```
> A2:=subs(k=-36,evalm(A));
```

$$A2 := \begin{bmatrix} -36 & 6 & 1 & -3 \\ 2 & -2 & -5 & 9 \\ 1 & -36 & 2 & -5 \\ 2 & 3 & -1 & 2 \end{bmatrix}$$

```
> A2b:=concat(A2,b);
```

$$\begin{bmatrix} -36 & 6 & 1 & -3 & 2 \\ 2 & -2 & -5 & 9 & 0 \\ 1 & -36 & 2 & -5 & e \\ 2 & 3 & -1 & 2 & e \end{bmatrix}$$

```
> gausselim(A2b);
```

$$\begin{bmatrix} 2 & -2 & -5 & 9 & 0 \\ 0 & 5 & 4 & -7 & e \\ 0 & 0 & -65 & 117 & 2+6e \\ 0 & 0 & 0 & 0 & 11e+1 \end{bmatrix}$$

que és compatible només per a $e = -\frac{1}{11}$. Resolent el sistema en aquest cas:

```
> b2:=subs(e=-1/11,evalm(b);
```

$$b2 := \left[2,\ 0,\ \frac{-1}{11},\ \frac{-1}{11}\right]$$

```
> linsolve(A2,b2);
```

$$\left[\_t_1 - \frac{8}{143},\ \_t_1,\ -45\,\_t_1 - \frac{5}{143},\ -25\,\_t_1 - \frac{1}{143}\right]$$

2.2 Operacions amb vectors i matrius

2.2.1 Espais vectorials. Canvi de base

Exemple 3 *Considerem els vectors de* $\mathbb{R}^4$, $\vec{u1} := (1,\ 1,\ 0,\ 0), \vec{u2} := (-3,\ 0,\ 1,\ 1), \vec{u3} := (-2,\ 1,\ 1,\ 1), \vec{u4} := (1,\ 1,\ 0,\ 0)$. *Trobeu una base del subespai que generen. Completeu-la fins a trobar una base* $\mathbb{R}^4$.

```
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

```
> u1:=vector([1,1,0,0]);u2:=vector([-3,0,1,1]); u3:=vector([-2,1,1,1]);
u4:=vector([1,1,0,0]);
```

$$u1 := [1,\ 1,\ 0,\ 0]$$

$$u2 := [-3,\ 0,\ 1,\ 1]$$

$$u3 := [-2,\ 1,\ 1,\ 1]$$

$$u4 := [1,\ 1,\ 0,\ 0]$$

La comanda **stackmatrix** permet definir la matriu dels vectors fila $u1, u2, u3, u4$

```
> A:=stackmatrix(u1,u2,u3,u4);
```

$$A := \begin{bmatrix} 1 & 1 & 0 & 0 \\ -3 & 0 & 1 & 1 \\ -2 & 1 & 1 & 1 \\ 1 & 1 & 0 & 0 \end{bmatrix}$$

```
> gausselim(A);
```

$$\begin{bmatrix} 1 & 1 & 0 & 0 \\ 0 & 3 & 1 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

El subespai generat per aquests quatre vectors té dimensió 2, i observem que els dos primers són linealment independents.
La instrucció **delrows** permet anul.lar files. En aquest cas des de la fila 3 fins la 4.

```
> B1:=delrows(A,3..4);
```

$$B1 := \begin{bmatrix} 1 & 1 & 0 & 0 \\ -3 & 0 & 1 & 1 \end{bmatrix}$$

Per a completar fins a tenir una base $\mathbb{R}^4$ hem de trobar 2 vectors independents amb els de la base del subespai. Provem amb $\vec{e_1}$ i $\vec{e_2}$ (observeu l'ús que es fa aquí de la instrucció **stackmatrix**)

```
> B2:=stackmatrix(B1,vector([1,0,0,0]),vector([0,1,0, 0]));
```

$$B2 := \begin{bmatrix} 1 & 1 & 0 & 0 \\ -3 & 0 & 1 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix}$$

```
> det(B2);
```

$$0$$

No són independents, n'agafem uns altres:

```
> B3:=stackmatrix(B1,vector([0,0,1,0]),vector([0,0,0, 1]));
```

$$B3 := \begin{bmatrix} 1 & 1 & 0 & 0 \\ -3 & 0 & 1 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> det(B3);
```

$$3$$

Els quatre vectors fila de la matriu $B3$, són la resposta d'aquest exercici.

Exemple 4 *Donats els vectors* $\mathcal{B} = \{\vec{u_1} = (-1,0,1,1,-1), \vec{u_2} = (0,1,-1,0,-1),$ $\vec{u_3} = (2,0,-1,0,1), \vec{u_4} = (1,0,2,-3,0), \vec{u_5} = (-2,1,0,0,4)\}$

a) Demostreu que $\mathcal{B}$ *és una base de* $\mathbb{R}^5$ *i escriviu la matriu de canvi de base de* $\mathcal{B}$ *a la base canònica.*

b) Trobeu les coordenades a la base $\mathcal{B}$ *del vector* $\vec{v} = (1,-1,-2,3,-4)$.

Els passos a fer amb *MAPLE V* són els següents:

```
> restart:
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
```

```
> u1:=vector([-1,0,1,1,-1]);u2:=vector([0,1,-1,0,-1]);
u3:=vector([2,0,- 1,0,1]);
u4:=vector([1,0,2,-3,0]);u5:=vector([-2,1,0,0,4]);
```

$$u1 := [-1, 0, 1, 1, -1]$$

$$u2 := [0, 1, -1, 0, -1]$$

$$u3 := [2, 0, -1, 0, 1]$$

$$u4 := [1, 0, 2, -3, 0]$$

$$u5 := [-2, 1, 0, 0, 4]$$

Per a obtenir la matriu C del canvi de la base $\mathcal{B}$ a la base usual, cal escriure en columna els vectors de la base $\mathcal{B}$, per a això es pot utilitzar la instrucció `concat`.

```
> C:=concat(u1,u2,u3,u4,u5);
```

$$C := \begin{bmatrix} -1 & 0 & 2 & 1 & -2 \\ 0 & 1 & 0 & 0 & 1 \\ 1 & -1 & -1 & 2 & 0 \\ 1 & 0 & 0 & -3 & 0 \\ -1 & -1 & 1 & 0 & 4 \end{bmatrix}$$

Comprovem que $\mathcal{B}$ és efectivament una base

```
> det(C);
```

$$48$$

Definim el vector v que està expressat en la base usual

```
> v:=vector([1,-1,-2,3,-4]);
```

$$v := [1, -1, -2, 3, -4]$$

per a expressar–lo en la base $\mathcal{B}$ s'ha de multiplicar la matriu inversa de C, que es calcula fent `inverse(C)`, pel vector

```
> evalm(inverse(C)&*v);
```

$$[0, 0, 0, -1, -1]$$

Aquesta última instrucció és equivalent a `multiply(inverse(C),v);`.

2.2.2 Aplicacions lineals

Exemple 5 *Considerem f un endomorfisme de $\mathbb{R}^5$ que té per matriu en la base canònica:*

$$A = \begin{pmatrix} \frac{-5}{2} & -8 & \frac{-17}{2} & \frac{-13}{2} & \frac{1}{2} \\ 1 & 2 & 2 & \frac{5}{3} & 0 \\ \frac{1}{3} & 2 & \frac{7}{3} & \frac{5}{3} & \frac{-1}{3} \\ 0 & 0 & 0 & 0 & 0 \\ \frac{23}{6} & 6 & \frac{35}{6} & \frac{31}{6} & \frac{1}{6} \end{pmatrix}$$

a) Busqueu una base i la dimensió de Im f.

b) Per quins valors de e i d, el vector $\vec{u} = (e, 1, 0, d, 1)$, pertany a Im f?

c) Pels valors de e i d que heu trobat en l'apartat anterior, busqueu $f^{-1}(\vec{u})$.

Introduïm primer la matriu de l'aplicació lineal.

```
> with(linalg):

Warning, new definition for norm
Warning, new definition for trace
```

Els vectors fe1,fe2,fe3,fe4,fe5, que són les columnes de la matriu A, són les imatges dels vectors de la base usual o canònica.

```
> fe1:=vector([-5/2,1,1/3,0,23/6]):fe2:=vector( [-8,2,2,0,6]):
fe3:=vector([-17/2,2,7/3,0,35/6]):
fe4:=vector([-13/2,5/3 ,5/3,0,31/6]):fe5:=vector([1/2,0,-1/3,0,1/6]):

> A:=concat(fe1,fe2,fe3,fe4,fe5);
```

$$A := \begin{bmatrix} \frac{-5}{2} & -8 & \frac{-17}{2} & \frac{-13}{2} & \frac{1}{2} \\ 1 & 2 & 2 & \frac{5}{3} & 0 \\ \frac{1}{3} & 2 & \frac{7}{3} & \frac{5}{3} & \frac{-1}{3} \\ 0 & 0 & 0 & 0 & 0 \\ \frac{23}{6} & 6 & \frac{35}{6} & \frac{31}{6} & \frac{1}{6} \end{bmatrix}$$

El nucli de l'aplicació lineal consta dels vectors u=(x,y,z,s,t) que tenen imatge 0. Els podem trobar resolent el sistema

```
> evalm(A &*[x,y,z,s,t]);
```

$$\left[-\frac{5}{2}x - 8y - \frac{17}{2}z - \frac{13}{2}s + \frac{1}{2}t,\ x + 2y + 2z + \frac{5}{3}s, \frac{1}{3}x + 2y + \frac{7}{3}z\frac{5}{3}s - \frac{1}{3}t,\ 0,\right.$$
$$\left. + \frac{23}{6}x + 6y + \frac{35}{6}z + \frac{31}{6}s + \frac{1}{6}t\right]$$

```
> solve({-5/2*x-8*y-17/2*z-13/2*s+1/2*t,x+2*y+ 2*z+5/3*s,
1/3*x+2*y+7/3*z+5/3*s-1/3*t,0,
23/6*x+6*y+35/6*z+31/6*s+1/6*t } ,{x,y,z,s,t});
```

$$\{s = -\frac{3}{2}y - \frac{3}{2}z,\ t = -y,\ x = \frac{1}{2}y + \frac{1}{2}z,\ z = z,\ y = y\}$$

Per tant un vector del nucli és de la forma $(\frac{1}{2}y + \frac{1}{2}z, y, z, -\frac{3}{2}y - \frac{3}{2}z, -y)$.

En el paquet **linalg** existeix la comanda **kernel(A)** que dóna directament una base del nucli

```
> kernel(A);
```

$$\{\left[\frac{1}{2}, 1, 0, \frac{-3}{2}, -1\right], \left[\frac{1}{2}, 0, 1, \frac{-3}{2}, 0\right]\}$$

Per a trobar una base de l'espai imatge, considerem la matriu B dels vectors $fe1 \ldots fe5$, i apliquem el mètode de reducció de Gauss. Obtenim que els vectors fe1,fe2,fe3 són linealment independents, així la dimensió de l'espai Imf és 3 com era d'esperar (per què?).

```
> B:=matrix(5,5,[fe1,fe2,fe3,fe4,fe5]):gausselim(B);
```

$$\begin{bmatrix} -8 & 2 & 2 & 0 & 6 \\ 0 & \frac{3}{8} & \frac{-7}{24} & 0 & \frac{47}{24} \\ 0 & 0 & \frac{1}{9} & 0 & \frac{1}{9} \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

El vector $u = (e, 1, 0, d, 1)$ serà de l'espai imatge, si pertany al subespai $< fe1, fe2, fe3 >$. O sigui, si la matriu Bim té rang 3.

```
> u:=vector([e,1,0,d,1]):Bim:=matrix(4,5,[fe1,fe2,fe3,u]):gausselim(Bim);
```

$$\begin{bmatrix} -8 & 2 & 2 & 0 & 6 \\ 0 & \frac{3}{8} & \frac{-7}{24} & 0 & \frac{47}{24} \\ 0 & 0 & \frac{1}{9} & 0 & \frac{1}{9} \\ 0 & 0 & 0 & d & -5-e \end{bmatrix}$$

Observem que el rang de Bim és 3 si $d = 0$ i $e = -5$. Trobem el(s) vector(s) que tenen per imatge $u = (-5, 1, 0, 0, 1)$.

Ja sabem que el sistema té dos graus de llibertat:

```
> uim:=subs(e=-5,d=0,evalm(u));
```

$$uim := [-5,\ 1,\ 0,\ 0,\ 1]$$

```
> linsolve(A,uim);
```

$$\left[-\frac{13}{2} + \frac{1}{2}\_t_1 + \frac{1}{2}\_t_2,\ \_t_1,\ \_t_2,\ -\frac{3}{2}\_t_1 - \frac{3}{2}\_t_2 + \frac{9}{2},\ -\_t_1 + 16\right]$$

Exemple 6 *Donat f un endomorfisme de $\mathbb{R}^5$ que té per matriu en la base canònica:*

$$A = \begin{pmatrix} \frac{1}{6} & 0 & \frac{1}{6} & \frac{-1}{6} & \frac{-1}{6} \\ \frac{-1}{24} & \frac{1}{4} & \frac{1}{12} & \frac{1}{24} & \frac{1}{6} \\ \frac{1}{6} & 1 & \frac{2}{3} & \frac{-1}{6} & \frac{1}{3} \\ \frac{-1}{4} & \frac{-3}{2} & -1 & \frac{1}{4} & \frac{-1}{2} \\ \frac{-1}{6} & 1 & \frac{1}{3} & \frac{1}{6} & \frac{2}{3} \end{pmatrix}$$

c) Busqueu una base i la dimensió dels subespais Nuc f i Im f.

b) Per quins valors de d i e, el vector $\vec{u} = (1, -1, -2, d, e)$, pertany a Im f?

c) Pels valors de d i e que heu trobat en l'apartat anterior, busqueu $f^{-1}(\vec{u})$.

d) Trobeu la matriu de f a la base $\mathcal{B}$ de l'Exemple 4.

e) Els vectors $\vec{u_1}$, $\vec{u_2}$ i $\vec{u_3}$ de l'Exemple 4, són del nucli? Per qué?

f) Té f vectors fixos, és a dir, vectors que compleixin $f(\vec{w}) = \vec{w}$? Busqueu-los.

Aplicarem un procediment semblant al de l'exemple anterior.

Introduïm primer la matriu de l'aplicació lineal.

```
> with(linalg):
> A:=matrix(5,5,[1/6,0,1/6,-1/6,-1/6,-1/24,1/4, 1/12,1/24,1/6,1/6,1,
2/3,-1/6,1/3,-1/4,-3/2,-1,1/4,-1/2,-1/6,1,1/3,1/6, 2/3]);
```

$$A := \begin{bmatrix} \frac{1}{6} & 0 & \frac{1}{6} & \frac{-1}{6} & \frac{-1}{6} \\ \frac{-1}{24} & \frac{1}{4} & \frac{1}{12} & \frac{1}{24} & \frac{1}{6} \\ \frac{1}{6} & 1 & \frac{2}{3} & \frac{-1}{6} & \frac{1}{3} \\ \frac{-1}{4} & \frac{-3}{2} & -1 & \frac{1}{4} & \frac{-1}{2} \\ \frac{-1}{6} & 1 & \frac{1}{3} & \frac{1}{6} & \frac{2}{3} \end{bmatrix}$$

Busquem una base del nucli:

```
> kernel(A);
```

$$\{[1, 0, 0, 1, 0], [2, 1, -2, 0, 0], [2, 0, -1, 0, 1]\}$$

Busquem una base de la imatge, aplicant el mètode de reducció de Gauss a la matriu que té per files les imatges dels vectors de la base usual que és la trasposta de A:

```
> gausselim(transpose(A));
```

$$\begin{bmatrix} \frac{1}{6} & \frac{-1}{24} & \frac{1}{6} & \frac{-1}{4} & \frac{-1}{6} \\ 0 & \frac{1}{4} & 1 & \frac{-3}{2} & 1 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$$

Observem que el rang de Imf és 2, una base està formada pels vectors:
$\{(4, -1, 4, -6, -4), (0, 1, 4, -6, 4)\}$.

Un vector pertany a Imf si és combinació lineal dels de la base. Ara volem trobar d i e perque el vector $(1, -1, -2, d, e)$ sigui de la imatge:

```
> matrix([[4, -1, 4, -6, -4], [0, 1, 4, -6, 4], [1, -1, -2, d, e]]);
```

$$\begin{bmatrix} 4 & -1 & 4 & -6 & -4 \\ 0 & 1 & 4 & -6 & 4 \\ 1 & -1 & -2 & d & e \end{bmatrix}$$

```
> gausselim(%);
```

$$\begin{bmatrix} 4 & -1 & 4 & -6 & -4 \\ 0 & 1 & 4 & -6 & 4 \\ 0 & 0 & 0 & d-3 & e+4 \end{bmatrix}$$

```
> solve({d-3, e+4},{d,e});
```

$$\{e = -4, d = 3\}$$

Busquem les antiimatges del vector $(1,-1,-2,3,-4)$:

```
> b:=vector([1, -1, -2, 3, -4]);
```

$$b := [1,\ -1,\ -2,\ 3,\ -4]$$

```
> linsolve(A,b);
```

$$[12+2\_t_2+2\_t_1+\_t_3,\ \_t_1,\ -6-\_t_2-2\_t_1,\ \_t_3,\ \_t_2]$$

Considerem la base $\mathcal{B}$ de l'Exemple 4, introduïm la matriu del canvi de base:

```
> C:=matrix(5,5,[-1,0,2,1,-2,0,1,0,0,1,1,-1,-1, 2,0,1,0,0,
-3,0,-1,-1,1,0,4]);
```

$$C := \begin{bmatrix} -1 & 0 & 2 & 1 & -2 \\ 0 & 1 & 0 & 0 & 1 \\ 1 & -1 & -1 & 2 & 0 \\ 1 & 0 & 0 & -3 & 0 \\ -1 & -1 & 1 & 0 & 4 \end{bmatrix}$$

Busquem la matriu de f en la base $\mathcal{B} = \{\vec{u_1}, \vec{u_2}, \vec{u_3}, \vec{u_4}, \vec{u_5}\}$.

```
> B:=evalm(inverse(C)&*A&*C);
```

$$B := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Observem que les tres primeres columnes d'aquesta matriu són 0, això significa que els 3 primers vectors de la base $\mathcal{B}$ són del nucli.

També podem observar que $f(\vec{u_4}) = \vec{u_4}$ (és un vector fix), busquem si hi ha més vectors fixos. Anomanem $I5$ a la matriu identitat

```
> I5 := evalm(array(identity, 1..5,1..5));
```

$$I5 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

i busquem els vectors que verifican $f(\vec{v}) = \vec{v}$ (és a dir $(A - I5)\vec{v} = \vec{0}$):

```
> kernel(A-I5);
```

$$\{[1, 0, 2, -3, 0]\}$$

2.3 Diagonalització d'endomorfismes

Exemple 7 *Sigui f un endomorfisme de $\mathbb{R}^5$ que té per matriu en la base canònica:*

$$A = \begin{pmatrix} 1 & -2 & -2 & -1 & 0 \\ \frac{1}{4} & \frac{1}{2} & 1 & \frac{3}{4} & \frac{1}{2} \\ \frac{-9}{4} & \frac{-5}{2} & -3 & \frac{-11}{4} & \frac{-1}{2} \\ 3 & 6 & 6 & 5 & 0 \\ \frac{-1}{4} & \frac{3}{2} & -1 & \frac{-3}{4} & \frac{3}{2} \end{pmatrix}$$

a) Escriviu el polinomi característic de f i factoritzeu-lo.

b) Busqueu els valors propis, amb la seva multiplicitat, i els vectors propis de f. És f diagonalitzable?

c) Demostreu que el següent conjunt de vectors:

$$\mathcal{B}' = \{(1,0,2,-3,0), (0,1,-1,0,-1), (-2,1,0,0,4), (-1,0,0,1,-1), (2,0,-1,0,1)\}$$

formen una base de $\mathbb{R}^5$ i escriviu la matriu de canvi de base de $\mathcal{B}'$ a la base canònica.

d) Busqueu la matriu de f a la base $\mathcal{B}'$.

Utilitzarem les comandes que té el paquet **linalg** per buscar polinomis característics, valors i vectors propis.

```
> with(linalg):
```

```
> A:=matrix(5,5,[1,-2,-2,-1,0,1/4,1/2,1,3/4,1/2 ,-9/4,-5/2,-3,-11/4,
-1/2,3,6,6,5,0,-1/4,3/2,-1,-3/4,3/2]);
```

$$A := \begin{bmatrix} 1 & -2 & -2 & -1 & 0 \\ \frac{1}{4} & \frac{1}{2} & 1 & \frac{3}{4} & \frac{1}{2} \\ \frac{-9}{4} & \frac{-5}{2} & -3 & \frac{-11}{4} & \frac{-1}{2} \\ 3 & 6 & 6 & 5 & 0 \\ \frac{-1}{4} & \frac{3}{2} & -1 & \frac{-3}{4} & \frac{3}{2} \end{bmatrix}$$

```
> charpoly(A,x);
```

$$x^5 - 5\,x^4 + 6\,x^3 + 4\,x^2 - 8\,x$$

```
> factor(%);
```

$$x\,(x+1)\,(x-2)^3$$

D'aquí podem deduir que f té el valor propi $\lambda_1 = 0$ amb multiplicitat $m_1 = 1$, el $\lambda_2 = -1$ amb multiplicitat $m_2 = 1$ i $\lambda_3 = 2$ amb multiplicitat $m_3 = 3$. Per buscar els vectors propis podem fer kernel$(A - \lambda_i I)$.

La comanda **eigenvalues(A)** dóna els valors propis de A i es veu també la multiplicitat:

```
> eigenvalues(A);
```

$$0,\ -1,\ 2,\ 2,\ 2$$

La comanda **eigenvectors(A)** dóna, per a cada valor propi, la seva multiplicitat i una base de vectors propis:

```
> eigenvectors(A);
```

$$[0,\ 1,\ \{\left[\frac{1}{2},\ 0,\ 1,\ \frac{-3}{2},\ 0\right]\}],\ [2,\ 3,\ \{[1,\ 0,\ -1,\ 1,\ 0],\ [2,\ 0,\ -1,\ 0,\ 1],\ [-10,\ 1,\ 4,\ 0,\ 0]\}],$$
$$[-1,\ 1,\ \{[0,\ -1,\ 1,\ 0,\ 1]\}]$$

f diagonalitza perquè tenim una base de $\mathbb{R}^5$ de vectors propis de f.

Introduïm els vectors de $\mathcal{B}'$:

```
> u1:=vector([1,0,2,-3,0]);u2:=vector([0,1,-1,0 ,1]);
u3:=vector([-2,1,0,0,4]); u4:=vector([-1,0,0,1,-1]);u5:=vector([2,0,-1,0,1]);
```

$$u1 := [1, 0, 2, -3, 0]$$

$$u2 := [0, 1, -1, 0, -1]$$

$$u3 := [-2, 1, 0, 0, 4]$$

$$u4 := [-1, 0, 0, 1, -1]$$

$$u5 := [2, 0, -1, 0, 1]$$

La matriu C del canvi de la base formada pels vectors $\mathcal{B}' = \{\vec{u_1}, \vec{u_2}, \vec{u_3}, \vec{u_4}, \vec{u_5}\}$ a la base usual és:

```
> C:=concat(u1,u2,u3,u4,u5);
```

$$C := \begin{bmatrix} 1 & 0 & -2 & -1 & 2 \\ 0 & 1 & 1 & 0 & 0 \\ 2 & -1 & 0 & 0 & -1 \\ -3 & 0 & 0 & 1 & 0 \\ 0 & -1 & 4 & -1 & 1 \end{bmatrix}$$

Fem el determinant per demostrar que els vectors de $\mathcal{B}'$ forman una base:

```
> det(C);
```

$$12$$

Busquem la matriu de f en aquesta base:

```
> Diag:=evalm(inverse(C)&*A&*C);
```

$$Diag := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 0 & 0 \\ 0 & 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 0 & 2 \end{bmatrix}$$

Observem que aquesta matriu té forma diagonal i per tant els vectors de la base $\mathcal{B}'$ són vectors propis.

Exemple 8 *Sigui f un endomorfisme de $\mathbb{R}^5$ que té per matriu en la base canònica:*

$$A = \begin{pmatrix} \frac{11}{6} & 1 & \frac{5}{6} & \frac{7}{6} & \frac{1}{6} \\ \frac{-1}{6} & 0 & \frac{-1}{6} & \frac{-1}{6} & \frac{1}{6} \\ \frac{1}{3} & 2 & \frac{7}{3} & \frac{5}{3} & \frac{-1}{3} \\ 0 & 0 & 0 & 0 & 0 \\ \frac{7}{6} & 3 & \frac{13}{6} & \frac{11}{6} & \frac{5}{6} \end{pmatrix}$$

a) Escriviu el polinomi característic de f.

b) Busqueu els valors propis i els vectors propis de f. És f diagonalitzable?

c) Demostreu que els vectors
$\mathcal{B} = \{(1,0,2,-3,0),(0,1,-1,0,-1),(-2,1,0,0,4),(-1,0,0,0,-1),(2,0,-1,0,1)\}$
formen una base de $\mathbb{R}^5$ i escriviu la matriu de canvi de base de $\mathcal{B}$ a la base canònica.

d) Busqueu la matriu de f a la base $\mathcal{B}$.

Fem el mateix que en l'Exemple 7.

```
> with(linalg):
> A:=matrix(5,5,[11/6, 1, 5/6, 7/6, 1/6, -1/6, 0, -1/6, -1/6, 1/6, 1/3,
2, 7/3,5/3, -1/3, 0, 0, 0, 0, 0, 7/6, 3, 13/6, 11/6, 5/6]);
```

$$A := \begin{bmatrix} \frac{11}{6} & 1 & \frac{5}{6} & \frac{7}{6} & \frac{1}{6} \\ \frac{-1}{6} & 0 & \frac{-1}{6} & \frac{-1}{6} & \frac{1}{6} \\ \frac{1}{3} & 2 & \frac{7}{3} & \frac{5}{3} & \frac{-1}{3} \\ 0 & 0 & 0 & 0 & 0 \\ \frac{7}{6} & 3 & \frac{13}{6} & \frac{11}{6} & \frac{5}{6} \end{bmatrix}$$

```
> charpoly(A,x);
```

$$x^5 - 5\,x^4 + 8\,x^3 - 4\,x^2$$

```
> factor(%);
```

$$x^2\,(x-1)\,(x-2)^2$$

```
> eigenvalues(A);
```

$$0,\ 0,\ 1,\ 2,\ 2$$

```
> eigenvectors(A);
```

$$[1,\ 1,\ \{[-2,\ 1,\ 0,\ 0,\ 4]\}],\ [0,\ 2,\ \{\left[\frac{-1}{2},\ -1,\ 0,\ \frac{3}{2},\ 1\right],\ \left[\frac{1}{2},\ 0,\ 1,\ \frac{-3}{2},\ 0\right]\}],\ [2,\ 2,\ \{[1,\ 0,\ 0,\ 0,\ 1]\}]$$

En aquest cas f no diagonalitza perquè el valor propi 2 té multiplicitat 2 i, en canvi, el subespai de vectors propis de valor propi 2 té dimensió 1.

```
> u1:=vector([1,0,2,-3,0]);u2:=vector([0,1,-1,0 ,-1]);
u3:=vector([-2,1,0,0,4]); u4:=vector([-1,0,0,0,-1]);u5:=vector([2,0,-1,0,1]);
```

$$u1 := [1,\ 0,\ 2,\ -3,\ 0]$$

$$u2 := [0,\ 1,\ -1,\ 0,\ -1]$$

$$u3 := [-2,\ 1,\ 0,\ 0,\ 4]$$

$$u4 := [-1,\ 0,\ 0,\ 0,\ -1]$$

$$u5 := [2,\ 0,\ -1,\ 0,\ 1]$$

```
> C:=concat(u1,u2,u3,u4,u5);
```

$$C := \begin{bmatrix} 1 & 0 & -2 & -1 & 2 \\ 0 & 1 & 1 & 0 & 0 \\ 2 & -1 & 0 & 0 & -1 \\ -3 & 0 & 0 & 0 & 0 \\ 0 & -1 & 4 & -1 & 1 \end{bmatrix}$$

```
> det(C);
```

$$-18$$

```
> B:=evalm(inverse(C)&*A&*C);
```

$$B := \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 0 & 2 \end{bmatrix}$$

2.4 Exercicis proposats.

1. Resoleu l'equació $((2,a,-1)-5(1,3,1))\cdot(1,-1,a)=-2$

2. Sigui $u=(1,1,0)$, $v=(0,1,1)$, $w=(1,-1,1)$ una base de $\mathbb{R}^3$ i sigui l'endomorfisme de $\mathbb{R}^3$ definit per $f(u)=u$, $f(v)=v$, $f(w)=-w$. Trobeu la matriu de f en la base usual de $\mathbb{R}^3$, així com les coordenades del vector $x=(1,1,1)$ en la base (u,v,w).

3. Discutiu i resoleu el sistema d'equacions:

$$\left\{\begin{array}{rl} \alpha x+y+z & =1, \\ x+\alpha\beta y+z & =\alpha, \\ x+\alpha y+z & =1, \end{array}\right.$$

   segons els valors dels paràmetres α i β.

4. Donada l'aplicació

$$f(x,y,z,t)=(x+y+z+t,x-y-z+t,x+t),$$

   trobeu les antiimatges del vector $(4,0,2)$, així com una base el nucli i l'espai imatge.

5. Sigui f un endomorfisme de $\mathbb{R}^5$ que té per matriu en la base canònica:

$$A=\begin{pmatrix} \frac{-7}{4} & \frac{-3}{2} & -1 & \frac{-5}{4} & \frac{-1}{2} \\ 0 & 0 & 0 & 0 & 0 \\ \frac{13}{12} & \frac{3}{2} & \frac{4}{3} & \frac{5}{4} & \frac{1}{6} \\ \frac{-5}{12} & \frac{-3}{2} & \frac{-5}{3} & \frac{-5}{4} & \frac{1}{6} \\ \frac{-2}{3} & 0 & \frac{1}{3} & 0 & \frac{-1}{3} \end{pmatrix}$$

a) Busqueu una base i la dimensió de Nuc f

b) Busqueu una base i la dimensió de Im f

c) Per quins valors de d i e, el vector $\vec{u} = (1, d, e, 1, 0)$, pertany a Im f?

d) Pels valors de d i e que heu trobat en l'apartat anterior, busqueu $f^{-1}(\vec{u})$

e) Demostreu que els vectors $\mathcal{B} = \{(1, 0, 2, -3, 0), (0, 1, -1, 0, -1), (-2, 1, 0, 0, 4), (-1, 0, 0, 0, -1), (2, 0, -1, 0, 1)\}$ formen una base de $\mathbb{R}^5$ i escriviu la matriu de canvi de base de $\mathcal{B}$ a la base canònica.

f) Trobeu les coordenades a la base $\mathcal{B}$ del vector $\vec{u} = (1, 0, -1, 1, 0)$

g) Calculeu la matriu de f a la base $\mathcal{B}$.

h) Busqueu les coordenades del vector $f(\vec{u})$ a la base canònica i a la base $\mathcal{B}$

i) Té f vectors que compleixin $f(\vec{w}) = -\vec{w}$?

j) Calculeu el polinomi característic i els valors i vectors propis de f.

k) Calculeu (Nuc $f)^{\perp}$

6. Estudieu si diagonalitzen els següents endomorfismes, i en cas afirmatiu obteniu les formes diagonals després de fer el canvi de base corresponent:

$$\begin{pmatrix} 3 & 2 & 3 \\ 2 & 6 & 6 \\ 5 & 10 & 17 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 \\ 1 & 1 & 0 \\ 1 & 1 & 2 \end{pmatrix}, \begin{pmatrix} -1 & -2 & 3 & 2 \\ 0 & 1 & 0 & 1 \\ -2 & -2 & 4 & 2 \\ 0 & 0 & 0 & 2 \end{pmatrix}.$$

7. Sigui l'endomorfisme $f : \mathbb{R}^4 \longrightarrow \mathbb{R}^4$ que té per matriu en la base canònica:

$$A = \begin{pmatrix} -7 & 0 & 10 & -5 \\ 11 & 4 & -31 & 20 \\ -15 & 0 & 28 & -15 \\ -20 & 0 & 40 & -22 \end{pmatrix}$$

a) Cerqueu els altres valors propis de l'endomorfisme.

b) Discutiu si l'endomorfisme diagonalitza.

c) Calculeu els vectors propis de f.

d) Quines són les coordenades del vector $(1, 2, 3, 4)$ en la base dels vectors propis?

e) Són propis, els vectors $w_1 = (36, 28, 18, 0)$ i $w_2 = (38, 15, 28, 18)$? Si és així, justifiqueu perquè no han sortit a la llista de vectors propis de l'apartat 7.

8. Determineu si les següents formes quadràtiques són definides positives, definides negatives o no definides.

$$q(x,y,z) = x^2 + xy + z^2 - 3xz - 2y^2$$
$$q(x,y,z) = 2x^2 - xy - zy - xz$$
$$q(x,y,z,t) = x^2 + xy - 3t^2 + tz + 5xz$$
$$q(x,y,z) = x^2 + y^2 + z^2 + 2zy + 2xy$$

9. Sigui

$$A = \frac{1}{4}\begin{pmatrix} 5 & \sqrt{6} & \sqrt{3} \\ \sqrt{6} & 4 & \sqrt{2} \\ \sqrt{3} & \sqrt{2} & 3 \end{pmatrix}$$

la matriu d'un endomorfisme simètric f a la base canònica de $\mathbb{R}^3$. Trobeu una base ortonormal orientada positivament, $\{\vec{v}_i\}$, en la qual f tingui una matriu diagonal. Escriviu en la base canònica i en la base $\{\vec{v}_i\}$ l'expressió de $q_f(\vec{u})$, essent q_f la forma quadràtica associada a f.

Lliçó 3

Eines bàsiques del càlcul

En aquest capítol exposem les utilitats del *MAPLE V* relacionades amb les lliçons 'Funcions reals d'una variable', 'Fórmula de Taylor' i 'Integració'.

A la introducció d'aquests apunts hi podem trobar un resum de les operacions elementals que necessitarem en aquest capítol; aquestes són:

- Aillar una variable (comanda `solve`). Exemple:

```
> x*y+x^2*y^2+x=0;
```

$$x\,y + x^2\,y^2 + x = 0$$

```
> solve(%,x);
```

$$0,\ -\frac{y+1}{y^2}$$

Ens ha donat les solucions $x = 0$ i $x = -(y+1)/y^2$

- Trobar les arrels d'un polinomi (comanda `roots(p)`). Exemple:

```
> p:=x^3-4*x^2+x+6;
```

$$p := x^3 - 4\,x^2 + x + 6$$

```
> roots(p);
```

$$[[3,\ 1],\ [-1,\ 1],\ [2,\ 1]]$$

Cada parella és una solució i la seva multiplicitat. Així doncs, les arrels són 3,-1,2 amb multiplicitat 1

- Factoritzar un polinomi (comanda `factor`). Recordem que només funciona en cas que les arrels siguin enteres o racionals. Exemple:

```
> factor(p);
```

$$(x-2)\,(x-3)\,(x+1)$$

- Expressió d'un nombre racional en decimal o fraccionari (comandes `digits` i `convert`). Exemple:

```
> convert(1.5,fraction);
```

$$\frac{3}{2}$$

- Operacions amb nombres complexos (la unitat imaginària es representa per `I`). Exemples:

```
> abs(3+4*I);
```

$$5$$

```
> argument(I);
```

$$\frac{1}{2}\pi$$

```
> Re(3+4*I);
```

$$3$$

```
> Im(3+4*I);
```

$$4$$

```
> conjugate(3+4*I);
```

$$3-4I$$

```
> solve(x^4-1=0);
```

$$1,\ -1,\ I,\ -I$$

3.1 Funcions reals d'una variable

3.1.1 Càlcul de límits

Per calcular el límit d'una funció f quan la variable x tendeix a a utilitzem la comanda `limit(f,x=a)`. Notem que a pot valdre infinit (`infinity`). Exemple:

```
> f:=(x^2-sqrt(x))/(sqrt(x)-1);
```

$$f := \frac{x^2-\sqrt{x}}{\sqrt{x}-1}$$

```
> limit(f,x=1);
```

$$3$$

Per tant el límit de f quan x tendeix a 1 és 3.

També podem calcular límits laterals afegint els arguments `left` i `right` a la comanda `lim`. Exemple:

```
> g:=(1+2^(1/x))/(1-2^(1/x));
```

$$g := \frac{1+2^{(\frac{1}{x})}}{1-2^{(\frac{1}{x})}}$$

```
> limit(g,x=0,left);
```

$$1$$

```
> limit(g,x=0,right);
```

$$-1$$

Els límits de la funció g per la esquerra i per la dreta són respectivament 1 i -1.

La funció sinus és periòdica, per tant no té límit en l'infinit, però sempre està acotada i això és el que obtenim al buscar el límit amb *MAPLE V*:

```
> limit(sin(x),x=-infinity);
```

$$-1..1$$

Per calcular el límit d'una succesió es fa servir la mateixa comanda que per una funció, però fent tendir la variable a infinit: `limit(f,n=infinity)`. Per exemple:

```
> f:=(n^4+n)^(1/3)/(n+1);
```

$$f := \frac{(n^4+n)^{1/3}}{n+1}$$

```
> limit(f,n=infinity);
```

$$\infty$$

3.1.2 Derivació

Per calcular la derivada d'una funció f respecte la variable x s'utilitza la comanda `diff(f,x)`. Exemple:

```
> f:=(sin(x))/(1+cos(x));
```

$$f := \frac{\sin(x)}{1+\cos(x)}$$

```
> diff(f,x);
```

$$\frac{\cos(x)}{1+\cos(x)} + \frac{\sin(x)^2}{(1+\cos(x))^2}$$

L' expressió anterior es pot simplificar de la manera següent

```
> simplify(%);
```

$$\frac{1}{1+\cos(x)}$$

Per obtenir les derivades segones i d'ordres superiors, afegim l'argument `$n` a la comanda `diff`.

Exemple 9 *Càlcul dels extrems de la funció $f(x) = x^3 - 2x + 1$.*

```
> f:=x^3-2*x+1;
```

$$f := x^3 - 2\,x + 1$$

Calculem la derivada d'$f(x)$:

```
> diff(f,x);
```

$$3\,x^2 - 2$$

A continuació cerquem els possibles extrems igualant a zero la derivada:

```
> solve(%);
```

$$\frac{1}{3}\sqrt{6},\ -\frac{1}{3}\sqrt{6}$$

Finalment calculem la segona derivada

```
> diff(f,x$2);
```

$$6\,x$$

i comprovem que en el primer punt crític és positiva i per tant correspon a un mínim, mentre que en el segon és negativa i per tant correspon a un màxim.

Exemple 10 *Sigui la funció* $f(x) = \begin{cases} e^{\frac{-1}{x^2}} & si\ x \neq 0 \\ 0 & si\ x = 0 \end{cases}$. *Calculeu:*

a) $\lim_{x\to 0+} f(x)$, $\lim_{x\to 0-} f(x)$, $\lim_{x\to 0} f(x)$. *És la funció continua en el 0?*

b) $\lim_{x\to\infty} f(x)$. *Té asímptota horitzontal? Quina?*

c) la primera derivada de f*: per* $x \neq 0$*, calculeu* $f'(x)$*; en* $x = 0$*, s'ha d'utilitzar la definició:* $f'(0) = \lim_{x\to 0} \dfrac{f(x) - f(0)}{x - 0}$. *És la funció derivable en el 0?*

d) Intervals de creixement/decreixement i extrems relatius.

e) La segona derivada de f.

f) Intervals de convexitat/concavitat i punts d'inflexió.

Intriduïm la funció:

```
> restart:
> f:=exp(-1/x^2);
```

$$f := e^{(-\frac{1}{x^2})}$$

Busquem el límit en el 0:

```
> limit(f,x=0,right);
```

$$0$$

```
> limit(f,x=0,left);
```

$$0$$

```
> limit(f,x=0);
```

$$0$$

En $x = 0$ la funció val 0 i és contínua però no es pot substituir el 0 directament a f tal com l'hem intriduït:

```
> subs(x=0,f);
Error, division by zero
```

Veiem que la funció té una asímptota horitzontal $y = 1$, perque el límit de la funció a l'infinit és 1:

```
> limit(f,x=infinity);
```

$$1$$

Busquem la derivada de f, quan $x \neq 0$ tenim

```
> derf:=diff(f,x);
```

$$derf := 2\,\frac{e^{(-\frac{1}{x^2})}}{x^3}$$

En $x = 0$ hem d'utilitzar la definició:

```
> limit(f/x,x=0);
```

$$0$$

Podem escriure la funció derivada: $derf(x) = \begin{cases} \frac{2}{x^3}e^{\frac{-1}{x^2}} & \text{si } x \neq 0 \\ 0 & \text{si } x = 0 \end{cases}$

Veiem que és contínua en el 0:

```
> limit(derf,x=0);
```

$$0$$

Tan sols s'anul.la en $x = 0$

```
> solve(derf,x);
```

La funció és decreixent en $(-\infty, 0)$, creixent en $(0, +\infty)$ i té un mínim en $x = 0$.

Busquem ara la segona derivada:

```
> diff(f,x$2);
```

$$-6\,\frac{e^{(-\frac{1}{x^2})}}{x^4} + 4\,\frac{e^{(-\frac{1}{x^2})}}{x^6}$$

```
> der2f:=simplify(%);
```

$$der2f := -2\,\frac{e^{(-\frac{1}{x^2})}\,(3\,x^2 - 2)}{x^6}$$

En $x = 0$ apliquem la definició:

```
> limit(derf/x,x=0);
```

$$0$$

La segona derivada és: $f(x) = \begin{cases} \frac{-2(3x^2-2)}{x^6} e^{\frac{-1}{x^2}} & \text{si } x \neq 0 \\ 0 & \text{si } x = 0 \end{cases}$

S'anul.la en $x = 0$ i en els punts:

```
>  solve(der2f,x);
```

$$\frac{1}{3}\sqrt{6},\ -\frac{1}{3}\sqrt{6}$$

Aquests dos zeros de la segona derivada són els punts d'inflexió de f. La funció té concavitat negativa en $(-\infty, -\sqrt{6}/3)$ i en $(\sqrt{6}/3, \infty)$ i concavitat positiva en $(-\sqrt{6}/3, \sqrt{6}/3)$.

Fem les gràfiques de la funció i de la seva derivada, observem que són contínues i derivables.

```
>  plot(f,x,colour=black);
```




```
>  plot(f,x=-0.5..0.5,colour=black);
```



```
> plot(derf,x,colour=black);
```



```
> plot(derf,x=-5..5,colour=black);
```



3.1.3 Gràfiques de funcions

Per dibuixar la gràfica d'una funció, utilitzarem la comanda `plot(f,x=a..b)`, el resultat és la representació gràfica de la funció y=f(x) en el interval [a,b].

```
> f:=x^2;
```

$$f := x^2$$

```
> plot(f,x=-5..5);
```




```
> plot(sin(x),x=-5..5);
```



La comanda plot també admet com a argument els valors que volem que surtin representats (variable y). Exemple:

```
> plot(sin(x),x=-5..5,y=-0.5..0.5);
```



Es poden representar simultàneament varies funcions, per exemple les funcions f, g i h, en un mateix interval [a,b] mitjançant `plot({f,g,h},x=a..b)`.

També podem dibuixar corbes en el pla donades parmètricament o en forma polar. Exemple:

```
> f:=cos(t),sin(t);
```

$$f := \cos(t),\, \sin(t)$$

```
> g:=1-sin(t),2+cos(t);
```

$$g := 1 - \sin(t),\, 2 + \cos(t)$$

```
> plot({[f,t=0..2*Pi],[g,t=0..2*Pi]});
```



L'argument **coords=polar** dintre de **plot** especifica que estem considerant coordenades polars.

Exemple 11 *Representem en el interval* $[-2\pi, 2\pi]$ *la funció* $r = 2sin(3a)$.

```
> plot(2*sin(3*a),a=-2*Pi..2*Pi,coords=polar);
```



Exemple 12 *Estudiem les funcions* $g(x) = \sin(1/x)$ *i* $f(x) = x\sin(1/x)$.

```
> restart;
> g:=sin(1/x);
```

$$g := \sin(\frac{1}{x})$$

```
> limit(g,x=0);
```

$$-1..1$$

Observem que en el 0 el límit d'aquesta funció no existeix, tenim una discontinuïtat essencial, però la funció sempre pren valors entre -1 i 1.

Tenim una asímptota horitzontal $y = 0$:

```
> limit(g,x=infinity);
```

$$0$$

Dibuixem la gràfica:

```
> plot(g,x);
```



Fem un dibuix on es vegi millor la funció prop de 0:

```
> plot(g,x=-1..1);
```



Considerem ara aquesta funció multiplicada per x:

```
> f:=x*g;
```

$$f := x\sin(\frac{1}{x})$$

En $x = 0$ el límit d'aquesta funció si que existeix:

```
> limit(f,x=0);
```

$$0$$

Tenim una asímptota horitzontal $y = 1$:

```
> limit(f,x=infinity);
```

$$1$$

```
> plot(f,x);
```



Dibuixem amb més detall la funció en l'interval $[-0.1, 1]$:

```
> plot(f,x=-0.1..1,y=-0.5..1,numpoints=7000);
```



En aquest dibuix no sembla que la funció sigui simètrica respecte l'eix d'ordenades, però sabem que si ho és:

```
> subs(x=-x,f);
```

$$-x\sin(-\frac{1}{x})$$

```
> simplify(");
```

$$x\sin(\frac{1}{x})$$

$f(-x) = f(x)$

Dibuixem un interval simètric:

```
> plot(f,x=-0.5..0.5);
```



És difícil apreciar bé la funció a l'entorn de 0, és necessari dibuixar molts punts:

```
> plot(f,x=-0.1..0.4,y=-0.25..0.25);
```



```
> plot(f,x=-0.2..0.2,y=-0.25..0.15, numpoints=8000,
colour=black,thickness=1);
```



Observem que hi ha molts extrems relatius

3.2 Desenvolupaments de Taylor

El desenvolupament de Taylor de grau n d'una funció f, n vegades derivable al voltant d'un punt a, es realitza amb la comanda `taylor(f,x=a;n)`. (Hem de tenir present que si el valor d'n és excessivament gran, trigarem molt a obtenir el resultat.)

Exemple 13 *Trobem el desenvolupament de Taylor de grau 6 de la funció* $f(x) = \frac{1}{x}$ *en un entorn del punt 1.*

```
> f:=1/x;
```

$$f := \frac{1}{x}$$

```
> taylor(f,x=1,6);
```

$$1 - (x-1) + (x-1)^2 - (x-1)^3 + (x-1)^4 - (x-1)^5 + \mathrm{O}((x-1)^6)$$

El resultat és un polinomi de grau 6 en potències de $x - 1$ més $O((x-1)^6)$, que ens indica l'ordre del residu. Utilitzant la comanda **convert** ens podem quedar amb l'aproximació polinómica de f.

```
> convert(%,polynom);
```

$$2 - x + (x-1)^2 - (x-1)^3 + (x-1)^4 - (x-1)^5$$

Finalment, amb la comanda **expand** podem obtenir l'aproximació de f en potències d' x.

```
> expand(%);
```

$$6 - 15\,x + 20\,x^2 - 15\,x^3 + 6\,x^4 - x^5$$

Quan la fórmula de Taylor es calcula a l'origen (fórmula de McLaurin), no és necessari especificar el segon argument. Exemple:

```
> f:=exp(x);
```

$$f := e^x$$

```
> taylor(f,x,5);
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \mathrm{O}(x^5)$$

Exemple 14 *Fent servir el desenvolupament de McLaurin (Taylor en l'origen) de la funció* $f(x) = \tan(x)$*, calculeu una aproximació de* $\tan(1)$*:*

a) Busqueu els polinomis de Taylor fins a grau 7.

b) Substituiu la x *per* 1.

Feu un dibuix de la funció i dels polinomis.

c) Calculeu ara els polinomis de graus 25, 27 y 29.

d) I el valor que prenen en $x = 1$.

Definim la funció i busquem els polinomis de Taylor de graus 1,3,5 i 7:

```
> f:=tan(x);
```

$$f := \tan(x)$$

```
> p1:=convert(taylor(f,x,3),polynom);
```

$$p1 := x$$

```
> p3:=convert(taylor(f,x,5),polynom);
```

$$p3 := x + \frac{1}{3}x^3$$

```
> p5:=convert(taylor(f,x,7),polynom);
```

$$p5 := x + \frac{1}{3}x^3 + \frac{2}{15}x^5$$

```
> p7:=convert(taylor(f,x,9),polynom);
```

$$p7 := x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7$$

Valorem els polinomis que hem trobat en $x = 1$:

```
> eval(subs(x=1,p1));
```

$$1$$

```
> evalf(subs(x=1,p3));
```

$$1.333333333$$

```
> evalf(subs(x=1,p5));
```

$$1.466666667$$

```
> evalf(subs(x=1,p7));
```

$$1.520634921$$

Els valors obtinguts són cada cop més propers entre si, observem les gràfiques:

```
> plot({f,p1,p3,p5,p7},x=-3..6,y=-10..10,colour=black);
```



```
> plot({f,p1,p3,p5,p7},x=0.5..1.5,y=1..2,linestyle=[0,1,2,3,4],
colour=black,thickness=1);
```



Busquem polinomis de Taylor de grau molt més elevat, valorem–los en $x = 1$ i observem que la diferència entre els dos últims valors obtinguts és menor que 0.000003

```
> p25:=convert(taylor(f,x,27),polynom):
> p27:=convert(taylor(f,x,29),polynom):
p29:=convert(taylor(f,x,31),polynom);
```

$$p29 := x + \frac{1}{3}x^3 + \frac{2}{15}x^5 + \frac{17}{315}x^7 + \frac{62}{2835}x^9 + \frac{1382}{155925}x^{11} + \frac{21844}{6081075}x^{13} + \frac{929569}{638512875}x^{15}$$

$$+\frac{6404582}{10854718875}x^{17}+\frac{443861162}{1856156927625}x^{19}+\frac{18888466084}{194896477400625}x^{21}$$
$$+\frac{113927491862}{2900518163668125}x^{23}+\frac{58870668456604}{369816065876859375}x^{25}+\frac{8374643517010684}{129805439119557764062}x^{27}$$
$$+\frac{689005380505609448}{263505041412702261046875}x^{29}$$

```
> evalf(subs(x=1,p25));
```

$$1.557396876$$

```
> evalf(subs(x=1,p27));
```

$$1.557403328$$

```
> evalf(subs(x=1,p29));
```

$$1.557405943$$

3.3 Integració

L'aplicació *MAPLE V* no tan sols calcula integrals definides, sinó que també pot trobar la primitiva d'una funció en termes de funcions conegudes, sempre que això sigui possible. Aquestes operacions es realitzen amb la comanda `int`.

Integrals indefinides (primitiva d'una funció).

La comanda `int(f,x)` calcula una primitiva de la funció f respecte la variable x. Exemple:

```
> f:=1/((1-x)*sqrt(1-x^2));
```

$$f := \frac{1}{(1-x)\sqrt{1-x^2}}$$

```
> int(f,x);
```

$$-\frac{\sqrt{-(-1+x)^2+2-2x}}{-1+x}$$

Podem simplificar l'expressió anterior amb la comanda `expand` o `simplify`.

```
> expand(%);
```

$$-\frac{\sqrt{1-x^2}}{-1+x}$$

Integrals definides

Per calcular una integral definida afegim a la comanda `int` un nou argument, que són els extrems d'integració. Així, `int(f,x=a..b)` calcula l'integral definida de f respecte x a l'intertval $[a, b]$ (si algún extrem d'integració és infinit es calcula la corresponent integral impròpia). Exemple:

```
> f:=x^2/(1+x^2);
```

$$f := \frac{x^2}{1+x^2}$$

```
> int(f,x=0..1);
```

$$1 - \frac{1}{4}\pi$$

Per calcular integrals dobles i triples reiterem la comanda `int`. Exemple:

```
> f:=y/(x^2+y^2);
```

$$f := \frac{y}{x^2+y^2}$$

```
> int(int(f,y=x..x*sqrt(2)),x=0..2);
```

$$\ln(3) - \ln(2)$$

Exemple 15 *Considerem la funció* $f(x) = \tan(x^2)$. *Volem calcular l'àrea limitada per la corba, l'eix OX i les rectes* $x_0 = 0$ *i* $x_1 = 1/2$.

a) Busqueu la integral definida de f *en l'interval* $[0, 1/2]$. *Calculeu de forma aproximada l'àrea.*

b) Utilitzeu els polinomis de Taylor de graus 2, 6 i 10 per calcular l'àrea.

Feu un dibuix aproximat de l'àrea buscada

Introduïm la funció:

```
> f:=tan(x^2);
```

$$f := \tan(x^2)$$

Quan busquem la integral definida no obtenim cap valor perque la comande `int` busca la primitiva de la funció i en aquest cas no existeix.

```
> int(f,x=0..1/2);
```

$$\int_0^{1/2} \tan(x^2)\, dx$$

Busquem l' àrea per mètodes numèrics:

```
> evalf(%);
```

$$.04204472108$$

Calculem ara alguns polinomis de Taylor de f i observem que al fer la integral definida d'aquests ens aproximem al valor obtingut numèricament:

```
> p2:=convert(taylor(f,x,6),polynom);
```

$$p2 := x^2$$

```
> int(p2,x=0..1/2);
```

$$\frac{1}{24}$$

```
> evalf(%);
```

$$.04166666667$$

```
> p6:=convert(taylor(f,x,10),polynom);
```

$$p6 := x^2 + \frac{1}{3}x^6$$

```
> int(p6,x=0..1/2);
```

$$\frac{113}{2688}$$

```
> evalf(%);
```

$$.04203869048$$

```
> p10:=convert(taylor(f,x,14),polynom);
```

$$p10 := x^2 + \frac{1}{3}x^6 + \frac{2}{15}x^{10}$$

```
> int(p10,x=0..1/2);
```

$$\frac{49727}{1182720}$$

```
> evalf(%);
```

$$.04204460904$$

```
> with(plots):
> implicitplot({y=f,y=p2,y=p6,y=p10,x=1/2},x=0..0.5,y=0..0.26,
colour=black);
```



3.4 Exercicis proposats

1. Treballeu amb el polinomi $x^3 + x + 1$: calculeu les seves arrels i descomposeu-lo en factors primers.

2. Calculeu les derivades d'ordre 1,2 i 3 de la funció $f(x) = \frac{\arccos x}{\sqrt{1-x^2}}$.

3. Una empresa porta a terme un estudi dels efectes de l'inversió en publicitat en les seves vendes, i observa que el nombre d'articles venuts, $A(d)n$ es relaciona amb la despesa publicitaria, d, a través de la funció $A(d) = \frac{150e^{d/20-1}}{e^{d/20+1}}$ També tenim les dades següents:

   - Preu de venda de cada article: 235.000
   - Cost de cada article: 12.000
   - Costos fixos totals: 2.000.000

   (a) Constuïu una funció $B(d)$ que dongui els beneficis en relació a la despesa publicitaria.

(b) Demostreu que $B(d)$ és una funció monòtona no decreixent i comproveu-ho gràficament.

(c) Quina és la despesa mínima en publicitat que cal fer per no tenir un balanç deficitari?

(d) Quin volum de beneficis s'aconseguiria si la despesa publicitaria fos il.limitada?

(e) Calculeu el limit de $B'(d)$ quan d tendeix a infinit, doneu una interpretació del resultat obtingut.

4. Calculeu i representeu gràficament els desenvolupaments de Taylor d'ordres 1,3, i 5 de la funció $f(x) = \sin(x)$. Observeu (dibuixant la gràfica) que a mesura que augmenta l'ordre hi ha més semblança amb la funció. Repetiu l'operació amb les funcions $\tan x$, $\arctan x$, i $\ln(1+x)$. Compareu els resultats.

5. Calculeu $\int_1^\infty \frac{\ln x}{x} dx$ i $\int_1^\infty \frac{\ln x}{x^2} dx$.

6. Feu la representació gràfica d'aquestes dues funcions que acabeu d'integrar per compendre millor el resultat obtingut.

7. Calculeu $\int \frac{e^{2x}}{(e^x-1)^{1/4}} dx$ i $\int \sqrt{x^2 - 6x - 7} dx$.

8. Representeu les corbes

$$x(t) = \frac{2t}{t^2+1}, \quad y(t) = \frac{t^2+1}{t^2-1}.$$

$$x(t) = \frac{t}{t^3+1}, \quad y(t) = \frac{t^2}{t^3+1}.$$

i les corbes en coordenades polars

$$r = 1 + \cos\theta$$

$$r = e^\theta$$

Lliçó 4

Corbes i superfícies. Extrems de funcions de vàries variables

4.1 Corbes a l'espai

En la secció de corbes a l'espai s'estudia primer la representació gràfica de corbes a l'espai amb l'objectiu de poder visualitzar una corba abans d'estudiar-ne les seves característiques, calcular-ne el Triedre de Frénet, ... Aquests càlculs es fan a la secció 4.1.2.

4.1.1 Corbes

Per a representar gràficament una corba es fa servir la instrucció `spacecurve`. Aquesta instrucció té com a paràmetres l'expressió parametritzada de la corba, vegeu l'exemple 16, o el nom que se li hagi donat, exemple 17. S'ha d'indicar també per a quins valors del paràmetre es vol representar la funció. És optatiu dir quants punts ha de tenir la corba, si hi volem eixos, ..., les diferents opcions estan explicades a `plot3d[option]` del menú **Help**.

Un cop obtingut el gràfic, si s'activa la finestra on hi ha el dibuix, apareix una caixa en lloc del dibuix. Si es mou el ratolí a l'hora que es prem el botó, s'obtenen diferents angles d'observació del gràfic, que torna a aparèixer en prémer **Return**.

Recordem que sempre que es treballa amb gràfics cal utilitzar el "paquet" `plots`

```
> with(plots):
```

Exemple 16 *Representeu gràficament la corba*

$$r(t) = (t\sin(t),\, t,\, t\cos(t))$$

```
> spacecurve([t*sin(t),t,t*cos(t)],t=-Pi..2*Pi,axes=normal) ;
```



Les corbes que es volen representar gràficament, es poden definir prèviament

Exemple 17

```
> knot:= [-10*cos(t)-2*cos(5*t)+ 15*sin(2*t),
-15*cos(2*t)+10*sin(t)-2*sin(5*t),10*cos(3*t),t=0..2*Pi]:
> spacecurve(knot, axes=frame);
```



El mateix gràfic s'hauria obtingut si en lloc de les dues instruccions anteriors haguéssim escrit:

```
> knot:=spacecurve([ -10*cos(t) - 2*cos(5*t) + 15*sin(2*t),
-15*cos(2*t) + 10*sin(t) - 2*sin(5*t), 10*cos(3*t)], :
```

```
t=0..2*Pi,axes=frame):
display(knot):
```

Observeu l'ús de la instrucció `display` que permet representar gràficament objectes gràfics definits prèviament.

Es poden dibuixar dues o més corbes simultàniament. En el següent exemple hi ha dibuixades dues hèlixs. Noteu que el sentit de la rotació és contrari.

```
> spacecurve({[3*cos(t),3*sin(t),0.2*t],
[4*cos(t),4*sin(t),-.2*t]}, t=0..6*Pi,axes=boxed);
```



4.1.2 Triedre de Frénet. Curvatura i torsió.

Exemple 18 *Trobeu la curvatura i la torsió de la corba*

$$r(t) = (e^t \cos t, e^t \sin t, e^{2t})$$

en el punt que correspon al valor del paràmetre $t = 0$.

```
> restart:with(linalg):
```

```
Warning, new definition for norm
Warning, new definition for trace
```

La primera instrucció inicialitza el sistema i la segona informa el programa que treballarem amb alguns "paquets" o "llibreries":

Per a definir una corba a l'espai escrivim la corba com a una funció vectorial de variable real

```
> corba1:=t->[exp(t)*cos(t),exp(t)*sin(t),exp(2 *t)];
```

$$corba1 := t \to [e^t \cos(t),\ e^t \sin(t),\ e^{(2\,t)}]$$

Es calculen els vectors derivada primera, segona i tercera

```
> fprim:=(t)->diff(corba1(t),t):fseg:=t->diff(corba1(t),t$2):
fter:=t->diff(corba1(t),t$3):
> fprim(t);fseg(t);fter(t);
```

$$[e^t \cos(t) - e^t \sin(t),\ e^t \sin(t) + e^t \cos(t),\ 2\,e^{(2\,t)}]$$

$$[-2\,e^t \sin(t),\ 2\,e^t \cos(t),\ 4\,e^{(2\,t)}]$$

$$[-2\,e^t \sin(t) - 2\,e^t \cos(t),\ 2\,e^t \cos(t) - 2\,e^t \sin(t),\ 8\,e^{(2\,t)}]$$

Per a calcular el vector tangent en $t = 0$, substituïm $t = 0$ en el vector derivada primera i obtenim *vtan0*, calculem la seva norma, i finalment tenim el *VtanUnit* el vector tangent unitari a la corba en $t = 0$.

```
> vtan0:=eval(subs(t=0,fprim(t)));Notan:=norm(vtan0,2);
VtanUnit:=evalm(vtan0/Notan);
```

$$vtan0 := [1,\ 1,\ 2]$$

$$Notan := \sqrt{6}$$

$$VtanUnit := \left[\frac{1}{6}\sqrt{6},\ \frac{1}{6}\sqrt{6},\ \frac{1}{3}\sqrt{6}\right]$$

El vector derivada segona

```
> vseg0:=eval(subs(t=0,fseg(t)));
```

$$vseg0 := [0,\ 2,\ 4]$$

permet calcular el vector binormal, que recordem és $b = \dfrac{r' \wedge r''}{\|r' \wedge r''\|}$ i en $t = 0$:

```
> bin0:=crossprod(vtan0,vseg0);Nobin:=norm(bin0 ,2);
VbinUnit:=evalm(bin0/Nobin);
```

$$bin0 := [0,\ -4,\ 2]$$

$$Nobin := 2\sqrt{5}$$

$$VbinUnit := \left[0,\ -\frac{2}{5}\sqrt{5},\ \frac{1}{5}\sqrt{5}\right]$$

La curvatura en el punt corresponent a $t = 0$ es calcula amb l'expressió
$\kappa = \dfrac{\|r' \wedge r''\|}{\|r'\|^3}$,

```
> kappa:=Nobin/Notan^3;
```

$$\kappa := \frac{1}{18}\sqrt{5}\sqrt{6}$$

El vector normal unitari, *VnorUnit* en $t = 0$

```
> VnorUnit:=crossprod(VbinUnit,VtanUnit);
```

$$VnorUnit := \left[-\frac{1}{6}\sqrt{5}\sqrt{6},\ \frac{1}{30}\sqrt{5}\sqrt{6},\ \frac{1}{15}\sqrt{5}\sqrt{6}\right]$$

I per a calcular la torsió $\tau = -\dfrac{(r' \wedge r'') \cdot r'''}{\|r' \wedge r''\|^2}$

```
> vter0:=eval(subs(t=0,fter(t)));
tor:=-dotprod( crossprod(vtan0,vseg0),vter0)/Nobin^2;
```

$$vter0 := [-2,\ 2,\ 8]$$

$$tor := \frac{-2}{5}$$

Observem que els vectors del Triedre de Frénet de la corba que estem estudiant són $\{VtanUnit,\ VnorUnit,\ VbinUnit\}$.

Exemple 19 *Donada la corba*

$$r(t) = (t, 1 + \frac{1}{t}, \frac{1}{t} - 1)$$

per a $t > 0$

a) Demostreu que la corba és plana i trobeu el pla que la conté.

b) Calculeu el Triedre de Frénet en $r(1)$ $(t = 1)$.

Com que la corba està definida per a valors positius del paràmetre es fa servir la instrucció `assume`. Noteu que per a designar que s'han restringit els valors de t, *MAPLE V* escriu $t\tilde{}$

```
> assume(t>0);corba2:=t->[t,1+1/t,1/t-1];
```

$$corba2 := t \rightarrow [t,\ 1+\frac{1}{t},\ \frac{1}{t}-1]$$

Els vectors derivada primera, segona i tercera

```
> fprim:=(t)->diff(corba2(t),t):fseg:=t->diff(corba2(t),t$2):
fter:=t->diff(corba2(t),t$3):
```

Calculem els vectors del Triedre de Frénet, *VtanUnit, VnorUnit, VbinUnit* en $t = 1$, com en l'exercici anterior.

```
> vtan0:=eval(subs(t=1,fprim(t)));
Notan:=norm(v tan0,2);VtanUnit:=evalm(vtan0/Notan);
```

$$vtan0 := [1,\ -1,\ -1]$$

$$Notan := \sqrt{3}$$

$$VtanUnit := \left[\frac{1}{3}\sqrt{3},\ -\frac{1}{3}\sqrt{3},\ -\frac{1}{3}\sqrt{3}\right]$$

```
> vseg0:=eval(subs(t=1,fseg(t)));
bin0:=crossprod(vtan0,vseg0);
Nobin:=norm(bin0,2);VbinUnit:=evalm(bin0/Nobin);
```

$$vseg0 := [0,\ 2,\ 2]$$

$$bin0 := [0,\ -2,\ 2]$$

$$Nobin := 2\sqrt{2}$$

$$VbinUnit := \left[0,\ -\frac{1}{2}\sqrt{2},\ \frac{1}{2}\sqrt{2}\right]$$

```
> VnorUnit:=crossprod(VbinUnit,VtanUnit);
```

$$VnorUnit := \left[\frac{1}{3}\sqrt{3}\sqrt{2},\ \frac{1}{6}\sqrt{3}\sqrt{2},\ \frac{1}{6}\sqrt{3}\sqrt{2}\right]$$

Els vectors del Triedre de Frénet en el punt que correspon al valor del paràmetre $t = 1$ són

```
> evalm(VtanUnit);evalm(VnorUnit);evalm(VbinUni t);
```

$$\left[\frac{1}{3}\sqrt{3},\ -\frac{1}{3}\sqrt{3},\ -\frac{1}{3}\sqrt{3}\right]$$

$$\left[\frac{1}{3}\sqrt{3}\sqrt{2},\ \frac{1}{6}\sqrt{3}\sqrt{2},\ \frac{1}{6}\sqrt{3}\sqrt{2}\right]$$

$$\left[0,\ -\frac{1}{2}\sqrt{2},\ \frac{1}{2}\sqrt{2}\right]$$

Perquè la corba sigui plana cal que el vector binormal en cada punt sigui constant.

Observeu que la instrucció `simplify` no pot actuar sobre un vector i s'ha de fer component a component, per aquest motiu el vector binormal en general es pot calcular

```
> nib:=crossprod(fprim(t),fseg(t));
nib:=[simplify(nib[1]/norm(nib,2)),
simplify(nib[2]/norm(nib,2)),
simplify(nib[3]/norm(nib,2))];
```

$$nib := \left[0,\ -\frac{2}{t^{\sim 3}},\ \frac{2}{t^{\sim 3}}\right]$$

$$nib := [0,\ -\frac{1}{2}\sqrt{2},\ \frac{1}{2}\sqrt{2}]$$

En efecte, el vector binormal és constant en cada punt de la corba. Per tant, la corba està continguda en un pla . Calculem un punt qualsevol de la corba , per exemple el que correspon a $t = 1$.

```
> punt:=eval(subs(t=1,corba2(t)));
```

$$punt := [1,\ 2,\ 0]$$

I busquem l'equació del pla que passa pel punt i té vector ortogonal **nib**

```
> dotprod(nib, [x-punt[1],y-punt[2],z-punt[3]])=0;
```

$$-\frac{1}{2}\sqrt{2}\,(y-2) + \frac{1}{2}\sqrt{2}\,z = 0$$

4.2 Representació gràfica de superfícies

4.2.1 Corbes de nivell

Donada una superfície que correspòn a la gràfica de la funció $z = f(x, y)$, en molts casos pot ser útil tenir-ne la representació en corbes de nivell. En *MAPLE V* aquesta representació s'obté amb la instrucció `contourplot`.

Exemple 20 *Dibuixeu les corbes de nivell de les funcions*

$$z = x^2 - y^2$$

$$z = x * \cos y$$

```
> contourplot(x^2-y^2,x=-3..3,y=-3..3);
contourplot(x*cos(y),x=-3..3,y=-2*Pi..2*Pi);
```





4.2.2 Superfícies

Per a representar gràficament una superfície es fan servir les instruccions **implicitplot3d**, o bé **plot3d** segons que la superfície estigui definida en forma implícita o explícita i paramètrica. Estudiarem un exemple de cada.

Exemple 21 *Representeu la superfície en forma implícita*

$$x^3 + y^3 + z^3 + 1 = (x + y + z + 1)^3$$

```
>  implicitplot3d( x^3+y^3+z^3+1 =(x+y+z+1)^3,x=-2..2,
y=-2..2, z=-2..2,grid=[13,13,13]);
```



Per a superfícies que vénen donades en forma paramètrica

```
> esfera:=(u,v)->[cos(v)*cos(u),cos(v)*sin(u),sin(v)]:
```

```
> plot3d(esfera(u,v),u=-Pi/2..Pi/2,
v=0..2*Pi,style=patchnogrid);
```



```
> mobius:=(u,v)->[(2-v*sin(u/2))*sin(u),
(2-v*sin(u/2))*cos(u), v*cos(u/2)]:
```

```
> plot3d(mobius(u,v),u=0..2*Pi,v=-1..1,
numpoints=3000);
```



I finalment la representació de superfícies expressades en forma explícita

```
> plot3d(x*exp(-x^2-y^2),x=-2..2,y=-2..2,grid=[50,50]);
```



Es pot respresentar simultàniament una corba i una superfície, amb la instrucció `display`.

Exemple 22 *Dibuixeu la corba de l'exemple 19 i el pla que la conté.*

```
> cor:=spacecurve(vpos,t=0.1..15,color=black,thickness=2):
pla:=implicitplot3d(z+2=y,x=-10..10,y=-10..10,z=-20..20,
color=green,grid=[3,3,3]):

> display(cor,pla);
```



4.3 Extrems de funcions de vàries variables

El càlcul dels extrems de funcions de vàries variables amb *MAPLE V* es pot fer estudiant la matriu hessiana en els punts que anul.len les derivades parcials de la funció o utilitzant comandes pròpies de *MAPLE V*. En els següents apartats s'estudia un exemple de cada situació.

4.3.1 Càlcul dels extrems relatius

Exemple 23 *Trobeu els extrems relatius de* $f(x,y) = 2x^2 + y^2 + 8x - 6y + 20$.

```
> restart:with(linalg):

Warning, new definition for norm
Warning, new definition for trace
```

Definim la funció de la qual volem buscar els extrems

```
> f:=(x,y)->2*x^2+y^2+8*x-6*y+20;
```

$$f := (x, y) \to 2\,x^2 + y^2 + 8\,x - 6\,y + 20$$

Calculem els punts crítics d'aquesta funció resolent el sistema d'equacions que s'obté a l'igualar a zero les derivades parcials.

```
> fx:=diff(f(x,y),x);fy:=diff(f(x,y),y);
```

$$fx := 4\,x + 8$$

$$fy := 2\,y - 6$$

```
> solve({fx=0,fy=0},{x,y});
```

$$\{x = -2, y = 3\}$$

Es calcula la matriu hessiana de la funció

```
> A:=hessian(f(x,y),[x,y]);
```

$$A := \begin{bmatrix} 4 & 0 \\ 0 & 2 \end{bmatrix}$$

que en aquest cas és constant–no depèn del punt. Com que la matriu A és diagonal, observem que els seus valors propis són $\lambda_1 = 4, \lambda_2 = 2$ que són tots dos positius i per tant A és definida positiva. En conseqüència, el punt $(-2, 3)$ correspon a un mínim de la funció f.

També es pot comprovar si A és definida (positiva o negativa) amb la instrucció:

```
> definite(A,'positive_def');
```

$$true$$

Com que el resultat d'aquesta última sentència és *true*, vol dir que la matriu obtinguda és definida positiva.

En el cas que el resultat hagués estat *false* s'hauria d'estudiar si la matriu hessiana és definida negativa amb la instrucció

```
> definite(A,'negative_def');
```

Si la matriu hessiana en el punt considerat no és ni definida positiva ni definida negativa, és que estem estudiant un punt singular que no correspon a un màxim ni un mínim de la funció.

En el cas que la matriu hessiana A no és constant, s'ha de substituir el punt (o punts) singular $(x0, y0)$ a la matriu A. Aquest procés es fa mitjançant la instrucció

```
> subs([x=x0,y=y0],eval(A));
```

Recordeu que es poden trobar directament els valors propis de la matriu A amb la instrucció

```
> eigenvals(A);
```

Una altra forma de resoldre problemes d'optimització és utilitzant la funció `extrema` que ha de ser carregada abans amb la instrucció `readlib(extrema)`[1].

Exemple 24 *Trobeu els extrems de la funció*

$$f(x, y) = x^3 + y^3 - 3 * x - 12 * y + 20$$

[1]Recordeu que en la versió 6 ja no es necessita la instrucció `readlib(extrema)`

```
> restart:readlib(extrema):
> f:=(x,y)->x^3+y^3-3*x-12*y+20;
```

$$f := (x, y) \to x^3 + y^3 - 3x - 12y + 20$$

```
> extrema(f(x,y),{},{x,y},'s');
```

$$2, 38$$

La funció **extrema** dóna el valor màxim i el mínim de la funció. Els punts crítics es guarden a la variable *'s'*.

```
> s;
```

$$\{\{x = 1, y = 2\}, \{x = 1, y = -2\}, \{x = -1, y = 2\}, \{x = -1, y = -2\}\}$$

Si s'avalua la funció en els punt crítics obtenim quin correspon a màxim i quin al mínim.

```
> f(1,2);f(1,-2);f(-1,2);f(-1,-2);
```

$$2$$

$$34$$

$$6$$

$$38$$

Observem que la funció pren el valor màxim, que és 38, en el punt $(-1, -2)$ i el mínim, que és 2, en $(1, 2)$. En els punts $(1, -2)$ i $(-1, 2)$ la funció pren els valors 34 i 6. Es pot comprovar que en cadascun d'aquests punts la matriu hessiana de f té un valor propi positiu i un de negatiu i per tant $(1, -2)$ i $(-1, 2)$ són punts sella de f .

4.3.2 Extrems de funcions amb restriccions

Exemple 25 *Trobeu els punts de l'esfera $(x-1)^2 + y^2 + (z-2)^2 = 27$ pels quals la suma de coordenades és màxima o mínima.*

Notem que la restricció igualada a 0 s'escriu entre claus $\{\}$. El demés és el mateix que en el cas sense restriccions.

```
> extrema(x+y+z,{(x-1)^2+y^2+(z-2)^2-27},{x,y,z},'s');
```

$$-6, 12$$

```
> s;
```

$$\{\{x = -2,\ z = -1,\ y = -3\},\ \{x = 4,\ y = 3,\ z = 5\}\}$$

Pel punt $(-2, -3, -1)$ la suma de les coordenades és mínima, val -6, mentre que en $(4, 3, 5)$ la suma de coordenades és màxima i val 12.

4.4 Exercicis proposats

1. Donada la corba
$$\alpha(t) = (\cos t + t \sin t, \sin t - t \cos t, 3)$$
   - Demostreu que la corba és plana i trobeu el pla que la conté.
   - Busqueu el triedre de Frénet d'aquesta corba per $t = \pi/2$
2. Representeu gràficament la corba anterior.
3. La corba
$$\alpha(t) = (a \cos t, a \sin t, kt)$$
   és l'hèlix circular.
   - Demostreu que el vector tangent en un punt qualsevol de l'hèlix, forma un angle constant amb l'eix OZ.
   - Comproveu que el vector binormal en un punt qualsevol de l'hèlix, forma un angle constant amb l'eix OZ.
4. Trobeu els extrems de la funció
$$f(x, y) = x^2 + xy + y^3 - y^2 - 3x - 2y + 1.$$
   Visualitzeu els extrems dibuixant la gràfica d'aquesta funció.
5. Trobeu els extrems relatius de la funció $f(x, y) = 2x^2 + y^2 + 8x - 6y + 20$.
6. Trobeu els semieixos de l'el.lipse $3x^2 + 3y^2 + 4xy = 1$.
7. Calculeu els extrems de la funció $f(x, y, z) = 2x+4y-z$, sobre l'esfera $x^2+y^2+z^2 = 9$. Digueu en quins punts f té un màxim i en quins un mínim.
8. Calculeu els extrems de la funció $f(x, y, z) = 2 * x^2 + y^2 + 3 * z^2$, sobre la superfície $2 * x - 3 * y - 4 * z = 49$.

9. Trobeu els punts de la corba $5x^2 + 6xy + 5y^2 = 8$ que estan a distància màxima i mínima de l'origen.

10. La funció $T(x, y) = 20 + 2x + 2y + z^2$ dóna la temperatura en cada punt de l'esfera d'equació $x^2 + y^2 + z^2 = 11$. Trobeu les temperatures extremes sobre la corba intersecció de l'esfera amb el pla $x + y + z = 3$

Índex temàtic

Bibliografia

[1] A. Carrillo de Albornoz et al. *MAPLE V. Aplicaciones matemticas para PC.* RA-MA Editorial. 1995.

[2] B.W. Char et al. *MAPLE V Library Reference Manual.* Springer-Verlag. 1991

[3] J. Getn, E. Pociello, J. Varea. *Problemas de matemticas aplicados a la economía y a la empresa.* Romanyà/Valls, s.a. 1994.